The Riemann Prime Counting Function $\Pi(n)$ and Generalized Divisor Sum Function $D_z(n)$

Nathan McKenzie
nathan@icecreambreakfast.com
December 1, 2013, Revised September 2014

Summary

This paper includes a number of results involving the Riemann's Prime counting function $\Pi(n)=\sum_{j=2}^{n}\frac{\Lambda(j)}{\log j}$. It includes two exact ways of expressing the difference between $\Pi(n)$ and $li(n)$ (the logarithmic integral), two novel methods for computing $\Pi(n)$ with favorable time/space bounds, a method for expressing $\Pi(n)$ as the roots of the generalized divisor summatory function $D_z(n)$, and a number of general results about $\Pi(n)$ and $D_z(n)$.

This paper is intended to be more explanatory than rigorous, so results will be explained but not proven. Instead, generally Mathematica code with provided with most assertions along with some arbitrarily chosen tests to demonstration the identities working as claimed.

The contents of the paper as as follows:

SECTION 1 (pg. 4) covers a handful of definitions used throughout the rest of the paper.

SECTION 2 (pg. 8) generalizes the Dirichlet Divisor problem for complex z as $D_z(n)$. Defining $D_{0,2}(n)=1$; $D_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor} D_{k-1,2}(\frac{n}{j})$ and $P_{0,2}(n)=1$; $P_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\frac{\Lambda(j)}{\log j}\cdot P_{k-1,2}(\frac{n}{j})$ and using binomial coefficients generalized to complex z, $\binom{z}{k}=\frac{z(z-1)...(z-k+1)}{k!}$, the two most important results here are

$$D_z(n)=\sum_{k=0}^{\lfloor\log_2 n\rfloor}\binom{z}{k}D_{k,2}(n)$$

$$D_z(n)=\sum_{k=0}^{\lfloor\log_2 n\rfloor}\frac{z^k}{k!}P_{k,2}(n)$$

SECTION 3 (pg. 13) gives identities for $\Pi(n)=\sum_{j=2}^{n}\frac{\Lambda(j)}{\log j}$, the Riemann Prime counting function, derivable from section 2's $D_z(n)$. Several results are listed, the most important being, with $D_{0,2}(n)=1$; $D_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor} D_{k-1,2}(\frac{n}{j})$ and $D_z(n)=\sum_{k=0}^{\lfloor\log_2 n\rfloor}\binom{z}{k}D_{k,2}(n)$,

$$\Pi(n)=\lim_{z\to 0}\frac{D_z(n)-1}{z}$$

$$\Pi(n)=\sum_{k=1}^{\lfloor \log_2 n\rfloor}\frac{(-1)^{k+1}}{k}D_{k,2}(n)$$

SECTION 4 (pg. 18) explores a method where, for some fixed value *n*, the roots of $D_z(n)=0$ can be used to express $D_z(n)$, Mertens function, and $\Pi(n)$. Similar arguments show how the zeros of a function related to $D_z(n)$ can express $\log(n!)$ and $\psi(n)$, the Chebyshev function. Defining $P_{0,2}(n)=1$; $P_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\frac{\Lambda(j)}{\log j}P_{k-1,2}(\frac{n}{j})$ and $D_z(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor}\frac{z^k}{k!}P_{k,2}(n)$, with some fixed *n,* there will be $\log_2 n$ values of *z* for which $D_z(n)=0$, and if we name those roots ρ, then

$$D_z(n)=\prod_{\rho}(1-\frac{z}{\rho})$$

$$\Pi(n)=-\sum_{\rho}\frac{1}{\rho}$$

SECTION 5 (pg. 26) uses an approach related to the Dirichlet hyperbola method to provide relatively fast ways of computing $D_{0,2}(n)=1$; $D_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}D_{k-1,2}(\frac{n}{j})$ and, hence, $D_z(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor}\binom{z}{k}D_{k,2}(n)$ and $\Pi(n)=\sum_{k=1}^{\lfloor \log_2 n\rfloor}\frac{(-1)^{k+1}}{k}D_{k,2}(n)$. The approach will compute values of $D_{k,2}(n)$ in better than $O(n)$ time and $O(\log n)$ space, with the core identity being

$$D_{k,a}(n)=\sum_{j=1}^{k}\binom{k}{j}\sum_{m=a}^{\lfloor n^{\frac{1}{k}}\rfloor}D_{k-j,m+1}(\frac{n}{m^j})$$
$$D_{1,a}(n)=\lfloor n\rfloor-a+1$$
$$D_{0,a}(n)=1$$

SECTION 6 (pg. 30) covers another, more complicated approach for computing $D_{k,2}(n)$, and, consequently, $D_z(n)$ and $\Pi(n)$, which will operate in, roughly, $O(n^{2/3}\log n)$ time and $O(n^{1/3}\log n)$ space. It is based in large measure on a generalization of an algorithm published by Marc Deléglise and Joöl Rivat in their paper “Computing the summation of the Mobius function” and is unrelated to the other main prime counting approaches.

SECTION 7 (pg. 35) shows one method for expressing the exact difference between $D_{0,2}(n)=1$; $D_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}D_{k-1,2}(\frac{n}{j})$ and $(-1)^k(1-\frac{\Gamma(k,-\log n)}{\Gamma(k)})$ where $\Gamma(k,-\log n)$ is the upper incomplete gamma function, then uses that approach to show that the difference between $\Pi(n)$ and

$li(n)$, the logarithmic integral, can be expressed as

$$D_{0,y}(x)=1;\ D_{k,y}(x)=\sum_{j=0}^{\lfloor x-y\rfloor} D_{k-1,y}\left(\frac{x}{j+y}\right)$$

$$\Pi(n)=li(n)-\log\log n-\gamma-\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{k}\int_{1}^{\infty}\frac{\partial}{\partial y}\left(y^{-k}D_{k,y+1}(n y^{k})\right)dy$$

A similar identity connecting the Chebyshev function $\psi(n)$ to n will also be given.

SECTION 8 (pg. 45) explores a Dirichlet convolution, partial sum equivalent of the relationship between the Riemann Zeta function $\zeta(s)$ and the Dirichlet eta function $\eta(s)$, then uses that relationship to show another expression for the relationship between $\Pi(n)$ and $li(n)$, the logarithmic integral. Given some positive whole number constant b , as the limit of b approaches infinity,

$$E_{0,2}(n)=1;\ E_{k,2}(n)=b^{-1}\sum_{j=b+1}^{\lfloor n\cdot b\rfloor}\left(b\cdot\left(\left\lfloor\frac{j}{b}\right\rfloor-\left\lfloor\frac{j-1}{b}\right\rfloor\right)-(b+1)\cdot\left(\left\lfloor\frac{j}{b+1}\right\rfloor-\left\lfloor\frac{j-1}{b+1}\right\rfloor\right)\right)E_{k-1,2}\left(\frac{n\cdot b}{j}\right)$$

$$\Pi(n)=li(n)-\log\log n-\gamma+\lim_{b\to\infty}\sum_{k=1}^{\left\lfloor\frac{\log n}{\log(b+1)-\log b}\right\rfloor}\frac{(-1)^{k-1}}{k}E_{k,2}(n)+\frac{1}{k}$$

A similar approach will then be taken to connect the Chebyshev function $\psi(n)$ to *n*.

Further Questions (pg. 55) lists a number of questions and issues left unresolved or opened up by the approaches in this paper.

1. Basics

Here we set up some notation, for Riemann's Prime Counting function, the divisor sum function, an adjusted divisor sum function, and an important prime power sum function.

The notation for Riemann's Prime counting function that we'll use is

$$\Pi(n) = \sum_{j=2}^{\lfloor n \rfloor} \frac{\Lambda(j)}{\log j} \tag{P1}$$

```
referenceRiemannPrimeCount[ n_ ] := Sum[ MangoldtLambda[ j ]/Log[ j ], { j, 2, n } ]
Table[ { n, FullSimplify[ referenceRiemannPrimeCount[ n ] ] }, { n, 1, 100 } ] // TableForm
```

http://mathworld.wolfram.com/RiemannPrimeCountingFunction.html

Here, $\Lambda(n)$ is the Von Mangoldt function.

http://mathworld.wolfram.com/MangoldtFunction.html

The divisor sum function is

$$D_0(n) = 1;\; D_k(n) = \sum_{j=1}^{\lfloor n \rfloor} D_{k-1}\left(\frac{n}{j}\right) \tag{1.1}$$

```
D1[ n_, 0 ] := 1
D1[ n_, k_ ] := Sum[ D1[ n / j, k - 1 ], { j, 1, Floor[ n ] } ]
Table[ D1[ n, k ], { n, 1, 50 }, { k, 1, 7 } ] // TableForm
```

http://en.wikipedia.org/wiki/Divisor_summatory_function

Examples of it include

$$D_1(n)=\sum_{j=1}^{\lfloor n\rfloor} 1 \qquad D_2(n)=\sum_{j=1}^{\lfloor n\rfloor}\sum_{k=1}^{\lfloor \frac{n}{j}\rfloor} 1 \qquad D_3(n)=\sum_{j=1}^{\lfloor n\rfloor}\sum_{k=1}^{\lfloor \frac{n}{j}\rfloor}\sum_{m=1}^{\lfloor \frac{n}{jk}\rfloor} 1$$

(1.2)

```
Dd1[ n_ ] := Sum[ 1, { j, 1, n } ]
Dd2[ n_ ] := Sum[ 1, { j, 1, n }, { k, 1, n/j } ]
Dd3[ n_ ] := Sum[ 1, { j, 1, n }, { k, 1, n/j }, { m, 1, n/(j k) } ]
D1[ n_, k_ ] := Sum[ D1[ n/j, k-1 ], { j, 1, Floor[ n ] } ]; D1[ n_, 0 ] := 1
Table[ { Dd1[ n ], "=", D1[ n, 1 ], Dd2[ n ], "=", D1[ n, 2 ], Dd3[ n ], "=", D1[ n, 3 ] }, { n, 1, 50 } ]//TableForm
```

It is related to the number of divisors function,

$$d_k(n)=\sum_{a_1\cdot a_2\cdot\ldots\cdot a_k=n} 1 \quad d_k(n)=\sum_{j|x} d_{k-1}(j) \quad d_1(n)=1 \quad d_0(n)=1\, if\ n=1, 0\ otherwise \quad d_k(n)=D_k(n)-D_k(n-1)$$

(1.3)

```
d1[ n_, k_ ] := Sum[ d1[ j, k-1 ] d1[ x/j, 1 ], { j, Divisors[ n ] } ];d1[ n_, 1 ] := 1;d1[ n_, 0 ] := 0;d1[ 1, 0 ] := 1
D1[ n_, 0 ] := 1
D1[ n_, k_ ] := Sum[ D1[ n/j, k-1 ], { j, 1, Floor[ n ] } ]
d1Alt[ n_, k_ ] := D1[ n, k ] - D1[ n-1, k ]
Grid[ Table[ { d1[ n, k ], d1Alt[ n, k ] }, { n, 1, 50 }, { k, 1, 7 } ] ]
```

A closely related function, which starts counting at 2 rather than 1, is

$$D_{0,2}(n)=1\,;\ D_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor} D_{k-1,2}\left(\frac{n}{j}\right)$$

(1.4)

```
D2[ n_, 0 ] := 1
D2[ n_, k_ ] := Sum[ D2[ n/j, k-1 ], { j, 2, Floor[ n ] } ]
Table[ D2[ n, k ], { n, 1, 50 }, { k, 1, 7 } ]//TableForm
```

with examples

$$D_{1,2}(n)=\sum_{j=2}^{\lfloor n\rfloor} 1 \qquad D_{2,2}(n)=\sum_{j=2}^{n}\sum_{k=2}^{\lfloor \frac{n}{j}\rfloor} 1 \qquad D_{3,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor \frac{n}{j}\rfloor}\sum_{m=2}^{\lfloor \frac{n}{jk}\rfloor} 1$$

(1.5)

```
D21[ n_ ] := Sum[ 1, { j, 2, n } ]
```

```
D22[ n_ ] := Sum[ 1, { j, 2, n }, { k, 2, n/j } ]
D23[ n_ ] := Sum[ 1, { j, 2, n }, { k, 2, n/j }, { m, 2, n/(j k) } ]
D2[ n_, k_ ] := Sum[ D2[ n/j, k-1 ], { j, 2, Floor[ n ] } ]; D2[ n_, 0 ] := 1
Table[ { D21[ n ], "=", D2[ n, 1 ], D22[ n ], "=", D2[ n, 2 ], D23[ n ], "=", D2[ n, 3 ] }, { n, 1, 50 } ]//TableForm
```

A final function we'll use is

$$P_{0,2}(n)=1;\ P_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\frac{\Lambda(j)}{\log j}P_{k-1,2}\left(\frac{n}{j}\right) \tag{1.6}$$

```
P2[ n_, 0 ] := 1
P2[ n_, k_ ] := Sum[ MangoldtLambda[ j ]/Log[ j ] P2[ n/j, k-1 ], { j, 2, Floor[ n ] } ]
Table[ FullSimplify[ P2[ n, k ] ], { n, 1, 50 }, { k, 1, 5 } ]//TableForm
```

with examples

$$P_{1,2}(n)=\Pi(n)=\sum_{j=2}^{\lfloor n\rfloor}\frac{\Lambda(j)}{\log j}\qquad P_{2,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor\frac{n}{j}\rfloor}\frac{\Lambda(j)}{\log j}\frac{\Lambda(k)}{\log k}\qquad P_{3,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor\frac{n}{j}\rfloor}\sum_{m=2}^{\lfloor\frac{n}{jk}\rfloor}\frac{\Lambda(j)}{\log j}\frac{\Lambda(k)}{\log k}\frac{\Lambda(m)}{\log m} \tag{1.7}$$

```
K[ n_ ] := FullSimplify[ MangoldtLambda[ n ]/Log[ n ] ]
P21[ n_ ] := Sum[ K[ j ], { j, 2, n } ]
P22[ n_ ] := Sum[ K[ j ] K[ k ], { j, 2, n }, { k, 2, n/j } ]
P23[ n_ ] := Sum[ K[ j ] K[ k ] K[ m ], { j, 2, n }, { k, 2, n/j }, { m, 2, n/(j k) } ]
P2[ n_, 0 ] := 1
P2[ n_, k_ ] := Sum[ K[ j ] P2[ n/j, k-1 ], { j, 2, Floor[ n ] } ]
Table[ { P21[ n ], "=", P2[ n, 1 ], P22[ n ], "=", P2[ n, 2 ], P23[ n ], "=", P2[ n, 3 ] }, { n, 1, 50 } ]//TableForm
```

A quick glance at the examples above should make clear that $P_{k,2}(n)=0$ when $n<2^k$, and $D_{k,2}(n)=0$ when $n<2^k$, but that $D_k(n)$ has no such property.

```
D1[ n_, k_ ] := Sum[ D1[ n / j, k - 1 ], { j, 1, Floor[ n ] } ];D1[ n_, 0 ] := 1
D2[ n_, k_ ] := Sum[ D2[ n/j, k-1 ], { j, 2, Floor[ n ] } ];D2[ n_, 0 ] := 1
P2[ n_, k_ ] := Sum[ FullSimplify[MangoldtLambda[ j ]/Log[ j ]] P2[ n/j, k-1 ], { j, 2, Floor[ n ] } ];P2[ n_, 0 ] := 1
Table[ {n, D2[ n, 4 ],D2[ n, 5 ], D2[ n, 6 ], P2[ n, 4 ],P2[ n, 5 ], P2[ n, 6 ], D1[ n, 4 ],D1[ n, 5 ], D1[ n, 6 ] }, { n, 1, 64 } ] // TableForm
```

When we use these first two functions in analogs to power series or polynomials, this property guarantees convergence after $\log_2 n$ terms, which is crucial.

Here's why we're defining these functions. For many identities, $\Pi(n)$, $D_k(n)$, $D_{k,2}(n)$, and

$P_{k,2}(n)$ relate to each other in ways that mirror $\log n$, n^k, $(n-1)^k$, and $(\log n)^k$, respectively.

2. Expanding the domain of the Divisor Summatory Function $D_z(n)$

This paper is all about treating the Divisor Summatory function, $D_z(n)$, as a function where z can be complex, continuous values. This section provides several identities for $D_z(n)$ with that property.

(1.1) defined $D_k(n)$ in such a way that *k* can be a whole number ≥ 0 . Everything changes if we extend the domain of *k* to continuous, complex arguments.

Here's one way to do that. Binomial coefficients can handle complex arguments when defined like so:

$$\binom{z}{k} = \frac{z(z-1)...(z-k+1)}{k!} \tag{2.1}$$

```
bin[ z_, k_ ] := Product[ z-j, { j, 0, k-1 } ]/k!
Grid[ Table[ { Binomial[ s + t I, u ], bin[ s + t I, u ] }, { s, -3, 3, 1.33 }, { t, -3, 3, 1.33 }, { u, 2, 4 } ] ]
```

Using (2.1), we can take an idea from pg. 421 of A. Ivic's "The Riemann Zeta-Function: Theory and Applications", and define $D_z(n)$ in the following way.

First, if we have some number *n* prime factored as $n = \prod_{p^\alpha | n} p^\alpha$, the number of divisors function $d_k(n)$ from (1.3) can be defined as

$$d_z(n) = \prod_{p^\alpha | n} (-1)^\alpha \binom{-z}{\alpha}$$

```
d1[ n_, k_ ] := Sum[ d1[ j, k-1 ] d1[ n/j, 1 ], { j, Divisors[ n ] } ];d1[ n_, 1 ] := 1;d1[ n_, 0 ] := 0;d1[ 1, 0 ] := 1
d1Alt[ n_, z_ ] := Product[ (-1)^p[ [ 2 ] ] Binomial[ -z, p[ [ 2 ] ] ], { p, FI[ n ] } ];FI[ n_ ] := FactorInteger[ n ];FI[ 1 ] := { }
Grid[ Table[ { d1[ n, k ], d1Alt[ n, k ] }, { n, 1, 100 }, { k, 1, 10 } ] ]
```

Sum that up and we have our Divisor Sum function that handles complex z.

$$D_z(n) = \sum_{j=1}^{\lfloor n \rfloor} d_z(j)$$

```
d1[ n_, z_ ] := Product[ (-1)^p[ [ 2 ] ] Binomial[ -z, p[ [ 2 ] ] ], { p, FI[ n ] } ];FI[ n_ ] := FactorInteger[ n ];FI[ 1 ] := { }
D1[ n_, z_ ] := Sum[ d1[ j, z ], { j, 1, n } ]
```

```
Grid[ Table[ D1[ 100, s+t I ], { s, -1.3, 4, .7 }, { t, -1.3, 4, .7 } ] ]
```

This works. We'll use it as our reference for later identities, in fact.

Despite this, we'll avoid $d_z(n)$ when possible - it's an erratic function and hard to reason about. Instead, using (2.1), we can also express $D_z(n)$ with $D_{0,2}(n)=1$; $D_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor} D_{k-1,2}(\frac{n}{j})$ from (1.4) as

$$D_z(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor} \binom{z}{k} D_{k,2}(n) \tag{D1}$$

```
d1[ n_, z_ ] := Product[ (-1)^p[ [ 2 ] ] Binomial[ -z, p[ [ 2 ] ] ], { p, FI[ n ] } ];FI[ n_ ] := FactorInteger[ n ];FI[ 1 ] := { }
D1[ n_, z_ ] := Sum[ d1[ j, z ], { j, 1, n } ]
D2[ n_, k_ ] := Sum[ D2[ n/j, k-1 ], { j, 2, Floor[ n ] } ]; D2[ n_, 0 ] := 1
D1Alt[ n_, z_ ] := Sum[ Binomial[ z, k ] D2[ n, k ], { k, 0, Log[ 2, n ] } ]
Grid[ Table[ { D1[ a=100, s+t I ], D1Alt[ a, s+t I ] }, { s, -1.3, 4, .7 }, { t, -1.3, 4, .7 } ] ]
```

Compare this to $n^z=\sum_{k=0}^{\infty} \binom{z}{k}(n-1)^k$

```
{ n^z, Sum[ Binomial[ z, k ] (n-1)^k, { k, 0, Infinity } ] }
```

Treating *z* as a complex, continuous variable opens up a lot of possibilities. It means that *z* can take negative values. We can apply limits to *z*. We can take derivatives with respect to it. (D1) might have zeros for *z* that contain valuable information. And so on.

It's particularly useful because $D_{-1}(n)$ is $M(n)$, the Mertens function, and $\lim_{z\to 0} \frac{D_z(n)-1}{z}$ is $\Pi(n)$, the Riemann Prime counting function. Hence, observations we make generally about $D_z(n)$ might have interesting implications for those two functions.

(D1) has a few variants. Here's another equation for $D_z(n)$, this time with $\mu(n)$ the Moebius function:

$$M_{0,2}(n)=1;\ M_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor} \mu(j) M_{k-1,2}(\frac{n}{j})$$

$$D_z(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor} \binom{-z}{k} M_{k,2}(n) \tag{D2}$$

```
referenced1[ n_, z_ ] := Product[ (-1)^p[ [ 2 ] ] Binomial[ -z, p[ [ 2 ] ] ], { p, FI[ n ] } ];FI[ n_ ] := FactorInteger[ n ];FI[ 1 ] := { }
referenceD1[ n_, z_ ] := Sum[ referenced1[ j, z ], { j, 1, n } ]
M2[ n_, k_ ] := Sum[ MoebiusMu[ j ] M2[ n/j, k-1 ], { j, 2, Floor[ n ] } ];M2[ n_, 0 ] := 1
D1Alt[ n_, z_ ] := Sum[ Binomial[ -z, k ] M2[ n, k ], { k, 0, Log[ 2, n ] } ]
Grid[ Table[ { referenceD1[ a=111, s+t I ], D1Alt[ a, s+t I ] }, { s, -1.3, 4, .7 }, { t, -1.3, 4, .7 } ] ]
```

Compare this to $n^z = \sum_{k=0}^{\infty} \binom{-z}{k} (n^{-1}-1)^k$

```
Table[ { n^z, Sum[ Binomial[ -z, k ] (n^(-1)-1)^k, { k, 0, Infinity } ] }, { z, -3, 10 } ]//TableForm
```

These ideas generalize to

$$A_{0,2}(n,a)=1;\ A_{k,2}(n,a)=\sum_{j=2}^{\lfloor n \rfloor} d_a(j)\, A_{k-1,2}\left(\frac{n}{j},a\right)$$

$$\boxed{D_z(n)=\sum_{k=0}^{\lfloor \log_2 n \rfloor} \binom{z/a}{k} A_{k,2}(n,a)}$$

(2.2)

```
referenced1[ n_, z_ ] := Product[ (-1)^p[ [ 2 ] ] Binomial[ -z, p[ [ 2 ] ] ], { p, FI[ n ] } ];FI[ n_ ] := FactorInteger[ n ];FI[ 1 ] := { }
referenceD1[ n_, z_ ] := Sum[ referenced1[ j, z ], { j, 1, n } ]
Aa[ n_, a_, k_ ] := Sum[ referenced1[ j, a ] Aa[ n/j, a, k-1 ], { j, 2, Floor[ n ] } ];Aa[ n_, a_, 0 ] := 1
D1Alt[ n_, z_, a_ ] := Sum[ Binomial[ z/a, k ] Aa[ n, a, k ], { k, 0, Log[ 2, n ] } ]
Grid[ Table[ { referenceD1[ b=111, s+2.3 I ], D1Alt[ b, s+2.3 I, t ] }, { s, -1.3, 4, .7 }, { t, -1.3, 4, .7 } ] ]
```

Compare this to $n^z = \sum_{k=0}^{\infty} \binom{z/a}{k} (n^a-1)^k$

```
Grid[ Table[ { n^z, Sum[ Binomial[ z/a, k ] (n^a-1)^k, { k, 0, Infinity } ] }, { z, -3, 10 }, { a, 1, 6 } ] ]
```

(D1), above, is one of two very important ways to express $D_z(n)$ with z a complex variable. The second way, in terms of $P_{0,2}(n)=1;\ P_{k,2}(n)=\sum_{j=2}^{\lfloor n \rfloor} \frac{\Lambda(j)}{\log j} \cdot P_{k-1,2}\left(\frac{n}{j}\right)$ from (1.6), is

$$\boxed{D_z(n)=\sum_{k=0}^{\lfloor \log_2 n \rfloor} \frac{z^k}{k!} P_{k,2}(n)}$$

(D3)

```
d1[ n_, z_ ] := Product[ (-1)^p[ [ 2 ] ] Binomial[ -z, p[ [ 2 ] ] ], { p, FI[ n ] } ];FI[ n_ ] := FactorInteger[ n ];FI[ 1 ] := { }
```

```
D1[ n_, z_ ] := Sum[ d1[ j, z ], { j, 1, n } ]
P2[ n_, k_ ] := Sum[ MangoldtLambda[ j ]/Log[ j ] P2[ n/j, k-1 ], { j, 2, Floor[ n ] } ];P2[ n_, 0 ] := 1
D1Alt[ n_, z_ ] := Sum[ z^k/k! P2[ n, k ], { k, 0, Log[ 2, n ] } ]
Grid[ Table[ { D1[ a=123, s+t I ], D1Alt[ a, s+t I ] }, { s, -1.5, 4, .7 }, { t, -1.1, 4, .7 } ] ]
```

Compare this to $n^z=\sum_{k=0}^{\infty}\frac{z^k}{k!}(\log n)^k$

```
{ n^z, Sum[ z^k/k! Log[ n ]^k, { k, 0, Infinity } ] }
```

Take a look at (D3). Because *z* is a complex, continuous variable, we can take the derivative of $D_z(n)$ with respect to *z*, giving us

$$\frac{\partial}{\partial z}D_z(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor-1}\frac{z^k}{k!}P_{k+1,2}(n) \tag{D4}$$

```
bin[z_,k_]:=Product[z-j,{j,0,k-1}]/k!
referenced1[n_,z_]:=Product[(-1)^p[[2]]bin[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1Alt[n_,z_]:=Sum[z^k/k! P2[n,k+1],{k,0,Log[2,n]-1}]
Table[{ Expand[D[referenceD1[n,z],z]],"=",D1Alt[n,z]},{n,1,100}]//TableForm
```

Compare this to $\frac{\partial}{\partial z}n^z=\sum_{k=0}^{\infty}\frac{z^k}{k!}(\log n)^{k+1}$

```
{D[n^z,z],Sum[ z^k/k! Log[n]^(k+1),{k,0,Infinity}]}
```

In fact, we can take its derivative an infinite number of times, although, as a polynomial of finite terms, most of those derivatives will be 0.

$$\frac{\partial^\alpha}{\partial z^\alpha}D_z(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor-\alpha}\frac{z^k}{k!}P_{k+\alpha,2}(n) \tag{D5}$$

```
bin[z_,k_]:=Product[z-j,{j,0,k-1}]/k!
referenced1[n_,z_]:=Product[(-1)^p[[2]]bin[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1Alt[n_,z_,a_]:=Sum[z^k/k! P2[n,k+a],{k,0,Log[2,n]-a}]
Grid[Table[{ Expand[D[referenceD1[n,z],{z,a}]],D1Alt[n,z,a]},{n,1,50},{a,0,5}]]
```

Compare this to $\frac{\partial^\alpha}{\partial z^\alpha} n^z = \sum_{k=0}^{\infty} \frac{z^k}{k!} (\log n)^{k+\alpha}$

```
Table[{D[n^z,{z,a}],Sum[ z^k/k! Log[n]^(k+a),{k,0,Infinity}]},{a,1,6}]//TableForm
```

Using these derivatives, $D_z(n)$, with respect to *z*, can be expressed as its Maclaurin series.

$$D_z(n) = \sum_{k=0}^{\lfloor \log_2 n \rfloor} \frac{z^k}{k!} \left(\frac{\partial^k}{\partial y^k} D_y(n) \text{ at } y=0 \right) \tag{D6}$$

```
bin[z_,k_]:=Product[z-j,{j,0,k-1}]/k!
referenced1[n_,z_]:=Product[(-1)^p[[2]]bin[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
D1Alt[n_, z_] :=Sum[ z^k/k! (D[referenceD1[n,y],{y,k}]/.y->0),{k,0,Log[2,n]}]
Grid[Table[{referenceD1[a=143,s+t I],D1Alt[a,s+t I]},{s,-1.5,4,.7},{t,-1.1,4,.7}]]
```

Compare this to $n^z = \sum_{k=0}^{\infty} \frac{z^k}{k!} \left(\frac{\partial^k}{\partial y^k} n^y \text{ at } y=0 \right)$

```
Grid[Table[{n^z,N[Sum[ z^k / k! (D[n^y,{y,k}]/.y->0),{k,0,50}]]},{n,-1.5,8},{z,1,5}]]
```

And it can be expressed as residues too.

$$D_z(n) = \sum_{k=0}^{\lfloor \log_2 n \rfloor} z^k \operatorname*{Res}_{m=0} \frac{D_m(n)}{m^{k+1}} \tag{D7}$$

```
bin[z_,k_]:=Product[z-j,{j,0,k-1}]/k!
referenced1[n_,z_]:=Product[(-1)^p[[2]]bin[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
Grid[Table[{referenceD1[ a = 155,s+t I],N[Sum[(s+t I)^k Residue[referenceD1[ a,m ]/(m^(k+1)),{m,0}],{k,0,50}]]},{s,-1.5,8},{t,1,5}]]
```

Compare this to $n^z = \sum_{k=0}^{\infty} z^k \operatorname*{Res}_{m=0} \frac{n^m}{m^{k+1}}$

```
Grid[Table[{n^z,N[Sum[ z^k Residue[ n^m/(m^(k+1)),{m,0}],{k,0,50}]]},{n,-1.5,8},{z,1,5}]]
```

3. Riemann's Prime Counting Function $\Pi(n)$

One of the main reasons to generalize $D_z(n)$ is its tight relationship to $\Pi(n)$, the Riemann Prime counting function. This section lists identities for $\Pi(n)$ based on that relationship.

Now that we have several ways to express $D_z(n)$ with *z* a complex continuous value, we can take limits with it.

And that lets us take the following limit, immediately giving us a very important expression for the Riemann Prime Counting function:

$$\Pi(x)=\lim_{z\to 0}\frac{D_z(n)-1}{z}$$

(P2)

```
referenceRiemanPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
referenced1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
Table[{referenceRiemanPrimeCount[n],"=",Limit[(referenceD1[n,z]-1)/z,z->0]},{n,1,100}]//TableForm
```

Compare to $\log n=\lim_{z\to 0}\frac{n^z-1}{z}$

```
{Log[n],Limit[ (n^z-1)/z,z->0]}
```

This same idea can also be expressed as

$$\Pi(n)=\sum_{k=1}^{\lfloor\log_2 n\rfloor}\frac{(-1)^{k+1}}{k}D_{k,2}(n)$$

(P3)

```
referenceRiemanPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
D2[n_,k_]:=Sum[D2[n/j,k-1],{j,2,Floor[n]}];D2[n_,0]:=1
RiemanPrimeCount[n_] := Sum[ (-1)^(k+1)/k D2[n,k],{k,1,Log[2,n]}]
Table[{referenceRiemanPrimeCount[n],"=",RiemanPrimeCount[n]},{n,1,100}]//TableForm
```

Compare to $\log n=\sum_{k=1}^{\infty}\frac{(-1)^{k+1}}{k}(n-1)^k$

```
{Log[n],Sum[(-1)^(k+1)/k (n-1)^k,{k,1,Infinity}]}
```

This is Linnik's identity summed, from pg. 343 of H. Iwaniec and E. Kowalski's “Analytic Number Theory”, more or less.

Other ways to arrive at Riemann's Prime counting function are

$$\Pi(n)=\frac{\partial}{\partial z}D_z(n)\, at\, z=0 \tag{P4}$$

```
referenceRiemanPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
bin[z_,k_]:=Product[z-j,{j,0,k-1}]/k!
referenced1[n_,z_]:=Product[(-1)^p[[2]]bin[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
Table[{referenceRiemanPrimeCount[n],"=",(Expand[D[referenceD1[n,z],z]]/.z->0)},{n,1,100}]//TableForm
```

Compare to $\log n=\frac{\partial}{\partial z}n^z\, at\, z=0$

```
{Log[n],D[n^z,z]/. z->0}
```

and

$$\Pi(n)=\operatorname*{Res}_{z=0}\frac{D_z(n)}{z^2} \tag{P5}$$

```
referenceRiemanPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
referenced1[n_,z_]:=Product[(-1)^p[[2]]Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
Table[{referenceRiemanPrimeCount[n],"=",Residue[referenceD1[n,z]/z^2,{z,0}]},{n,1,100}]//TableForm
```

Compare to $\log n=\operatorname*{Res}_{z=0}\frac{n^z}{z^2}$

```
{Log[n],Residue[ n^z/z^2,{z,0}]}
```

Remembering our examples of $D_{k,2}(n)$ from (1.5), (P3) can be written more explicitly in sum notation as

$$\Pi(n)=\sum_{j=2}^{\lfloor n\rfloor}1-\frac{1}{2}\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor\frac{n}{j}\rfloor}1+\frac{1}{3}\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor\frac{n}{j}\rfloor}\sum_{l=2}^{\lfloor\frac{n}{j\cdot k}\rfloor}1-\frac{1}{4}\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor\frac{n}{j}\rfloor}\sum_{l=2}^{\lfloor\frac{n}{j\cdot k}\rfloor}\sum_{m=2}^{\lfloor\frac{n}{j\cdot k\cdot l}\rfloor}1+\frac{1}{5}\ldots$$

(P5)

```
referenceRiemanPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
(* PAlt is trunctated and stops working after n=2^6-1*)
P2Alt[n_] := Sum[ 1,{j,2,n}]-
1/2 Sum[ 1,{j,2,n},{k,2,n/j}]+
1/3 Sum[1,{j,2,n},{k,2,n/j},{l,2,n/(j k)}]-
1/4 Sum[1,{j,2,n},{k,2,n/j},{l,2,n/(j k)},{m,2,n/(j k l)}]+
1/5 Sum[1,{j,2,n},{k,2,n/j},{l,2,n/(j k)},{m,2,n/(j k l)},{o,2,n/(j k l m)}]
Table[{referenceRiemanPrimeCount[n],"=",P2Alt[n]},{n,1,63}]//TableForm
```

The core idea here can be rewritten recursively as

$$F_k(n)=\sum_{j=2}^{\lfloor n\rfloor}\frac{1}{k}-F_{k+1}(\lfloor\frac{n}{j}\rfloor)$$
$$\Pi(n)=F_1(n)$$

(P6)

```
referenceRiemanPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
Ff[n_,k_] := Sum[ 1/k-Ff[Floor[n/j],k+1],{j,2,n}]
Table[{referenceRiemanPrimeCount[n],"=",Ff[n,1]},{n,1,100}]//TableForm
```

or as

$$F_k(n,j)=\frac{1}{k}-F_{k+1}(\lfloor\frac{n}{j}\rfloor,\lfloor\frac{n}{j}\rfloor)+F_k(n,j-1)$$
$$F_k(n,1)=0$$
$$\Pi(n)=F_1(n,n)$$

(P7)

```
referenceRiemanPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
Ff[n_,j_,k_] := 1/k - Ff[Floor[n/j],Floor[n/j],k+1]+Ff[n,j-1,k]
Ff[n_,1,k_]=0
Table[{referenceRiemanPrimeCount[n],"=",Fe[n,n,1]},{n,1,100}]//TableForm
```

A slight variant of (P5) is

$$\Pi(n)=z^{-1}(\sum_{j=2}^{\lfloor n\rfloor}d_z(j)-\frac{1}{2}\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor\frac{n}{j}\rfloor}d_z(j)d_z(k)+\frac{1}{3}\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor\frac{n}{j}\rfloor}\sum_{l=2}^{\lfloor\frac{n}{j\cdot k}\rfloor}d_z(j)d_z(k)d_z(l)-\frac{1}{4}\ldots)$$

(P8)

Compare to $\log n = z^{-1}\left(\frac{(n^z-1)}{1} - \frac{(n^z-1)^2}{2} + \frac{(n^z-1)^3}{3} - \frac{(n^z-1)^4}{4} + \frac{(n^z-1)^5}{5} \ldots\right)$

```
{Log[n],Sum[ (-1)^(k-1)/k (n^z-1)^{k},{k,1,Infinity}]/z}
```

Generalizing, $P_{0,2}(n)=1;\ P_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor} \frac{\Lambda(j)}{\log j} \cdot P_{k-1,2}\left(\frac{n}{j}\right)$ from (1.6) also has a few useful identities. It can be expressed in terms of $D_{k,2}(n)$ as

$$P_{j,2}(n) = \sum_{k=0}^{\lfloor \log_2 n \rfloor} \frac{1}{k!}\left(\frac{\partial^k}{\partial y^k}(\log(1+y))^j \text{ at } y=0\right) \cdot D_{k,2}(n) \tag{3.1}$$

```
P2[n_,k_]:=Sum[MangoldtLambda[j]/Log[j] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D2[n_,k_]:=Sum[D2[n/j,k-1],{j,2,Floor[n]}];D2[n_,0]:=1
P2Alt[n_, j_] := Sum[ 1/k!(D[Log[1+x]^j,{x,k}]/.x->0) D2[n,k],{k,0,Log[2,n]}]
Table[FullSimplify[{P2[n,k], P2Alt[n,k]}],{n,1,50},{k,1,5}]//TableForm
```

Compare to $(\log n)^j = \sum_{k=0}^{\infty} \frac{1}{k!}\left(\frac{\partial^k}{\partial y^k}(\log(1+y))^j \text{ at } y=0\right) \cdot (n-1)^k$

```
Table[{Table[ 1/k! D[Log[1+n]^j,{n,k}]/.n->0,{k,0,14}],CoefficientList[Series[Log[1+n]^j,{n,0,14}],n]},{j,1,7}]
```

It can be expressed as the derivative of $D_z(n)$ as

$$P_{k,2}(n) = \frac{\partial^k}{\partial z^k} D_z(n)\, at\ z=0 \tag{3.2}$$

```
P2[n_,k_]:=Sum[MangoldtLambda[j]/Log[j] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
bin[z_,k_]:=Product[z-j,{j,0,k-1}]/k!
referenced1[n_,z_]:=Product[(-1)^p[[2]]bin[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
Grid[Table[{FullSimplify[P2[n,k]],"=",(D[referenceD1[n,z],{z,k}]/.z->0)},{n,1,50},{k,1,5}]]
```

Compare to $(\log n)^j = \frac{\partial^k}{\partial z^k} n^z\, at\ z=0$

```
{Log[n]^j,D[n^z,z]^j/. z->0}
```

And it can be expressed as a residue as

$$P_{k,2}(n) = k!\operatorname*{Res}_{z=0} \frac{D_z(n)}{z^{k+1}} \tag{3.3}$$

```
P2[n_,k_]:=Sum[MangoldtLambda[j]/Log[j] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
bin[z_,k_]:=Product[z-j,{j,0,k-1}]/k!
referenced1[n_,z_]:=Product[(-1)^p[[2]]bin[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
Grid[Table[{FullSimplify[P2[n,k]],"=",k! Residue[referenceD1[n,z]/z^(k+1),{z,0}]},{n,1,50},{k,1,5}]]
```

Compare to $(\log n)^k = k!\operatorname*{Res}_{z=0} \frac{n^z}{z^{k+1}}$

```
Table[{Log[n]^k,k! Residue[n^z/z^(k+1),{z,0}]},{k,1,10}]//TableForm
```

4. The Roots of $D_z(n)$

Starting with $D_z(n)$ and using n a fixed value and z a complex variable, this section explores the roots of $D_z(x)=0$ and shows how they can be used in expressions for $D_z(n)$, $\Pi(n)$, and the Mertens function. It then presents several other similar sets of zeros, including one for the Chebyshev function $\psi(n)$.

Given we can define the divisor sum function for complex z $D_z(n)$ using $P_{k,2}(n)$ (from (1.6)) as

$$P_{0,2}(n)=1;\ P_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\frac{\Lambda(j)}{\log j}P_{k-1,2}\left(\frac{n}{j}\right) \text{ and } D_z(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor}\frac{z^k}{k!}P_{k,2}(n)$$

```
P2[n_,k_]:=Sum[MangoldtLambda[j]/Log[j] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1[n_,z_]:=Sum[z^k/k! P2[n,k],{k,0,Log[2,n]}]
```

and given that $P_{k,2}(n)=0$ when $n<2^k$,

then for some fixed n, $D_z(n)$ can be treated as a polynomial of degree $\log_2 n$ with z as its variable. As an example,

$$D_z(100)=\sum_{k=0}^{\lfloor \log_2 100\rfloor}\frac{z^k}{k!}P_{k,2}(100)=1+\frac{428}{15}z+\frac{16289}{360}z^2+\frac{331}{16}z^3+\frac{611}{144}z^4+\frac{67}{240}z^5+\frac{7}{720}z^6$$

```
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1[n_,z_]:=Sum[z^k/k! P2[n,k],{k,0,Log[2,n]}]
D1[100,z]
```

Thus, it should have $\log_2 n$ solutions for z where $D_z(n)=0$.

```
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1[n_,z_]:=Sum[z^k/k! P2[n,k],{k,0,Log[2,n]}]
Table[{n,Roots[ D1[n,z]==0,z]},{n,2,31}]//TableForm
```

Denote the roots ρ, and, through a bit of algebraic manipulation, and because $D_0(n)=1$ and $D_1(n)=\lfloor n\rfloor$, we have

$$D_z(n)=\prod_\rho \left(1-\frac{z}{\rho}\right)$$

(D8)

```
referenced1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1[n_,z_]:=Sum[z^k/k! P2[n,k],{k,0,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1[n,z]==0,z][[All,2]]
D1Alt[n_, z_] := Product[ 1-z/r,{r,zeros[n]}]
Grid[Table[{referenceD1[a=111,s+t I],D1Alt[a,s+t I]},{s,-1.3,4,.7},{t,-1.3,4,.7}]]
```

and

$$D_z(n)=\lfloor n\rfloor\cdot\prod_\rho \left(1-\frac{z-1}{\rho-1}\right)$$

(D9)

```
referenced1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1[n_,z_]:=Sum[z^k/k! P2[n,k],{k,0,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1[n,z]==0,z][[All,2]]
D1Alt[n_, z_] := n Product[ 1-(z-1)/(r-1),{r,zeros[n]}]
Grid[Table[{referenceD1[a=131,s+t I],D1Alt[a,s+t I]},{s,-1.1,4,.7},{t,-1.4,4,.7}]]
```

These zeros are connected to $\Pi(x)$, the Riemann Prime counting function, as

$$\Pi(n)=-\sum_\rho \frac{1}{\rho}$$

(P9)

```
ReferenceRiemannPrimeCnt[n_]:=Sum[MangoldtLambda[j]/Log[j],{j,2,n}]
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1[n_,z_]:=Sum[z^k/k! P2[n,k],{k,0,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1[n,z]==0,z][[All,2]]
P21Alt[n_] := -Sum[ 1/r,{r,zeros[n]}]
Table[{N[ReferenceRiemannPrimeCnt[n]],P21Alt[n]},{n,4,100}]//TableForm
```

Here are specific results with these roots, with $M(n)$ Mertens function and $D(n)$ the standard Dirichlet Divisor problem ($D_2(n)$ in this paper)

$$M(n)=\prod_\rho \left(1+\frac{1}{\rho}\right)$$

$$1=\prod_{\rho}\left(1-\frac{0}{\rho}\right)$$

$$\lfloor n\rfloor=\prod_{\rho}\left(1-\frac{1}{\rho}\right)$$

$$D(n)=\prod_{\rho}\left(1-\frac{2}{\rho}\right) \tag{4.1}$$

```
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1[n_,z_]:=Sum[z^k/k! P2[n,k],{k,0,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1[n,z]==0,z][[All,2]]
Table[{Sum[MoebiusMu[j],{j,1,n}],Product[ 1+1/r,{r,zeros[n]}]},{n,4,100}]//TableForm
Table[{1,Product[ 1-0/r,{r,zeros[n]}]},{n,4,100}]//TableForm
Table[{n,Product[ 1-1/r,{r,zeros[n]}]},{n,4,100}]//TableForm
Table[{Sum[1,{j,1,n},{k,1,Floor[n/j]}],Product[ 1-2/r,{r,zeros[n]}]},{n,4,100}]//TableForm
```

and

$$M(n)=n\cdot\prod_{\rho}\left(1+\frac{2}{\rho-1}\right)$$

$$1=\lfloor n\rfloor\cdot\prod_{\rho}\left(1+\frac{1}{\rho-1}\right)$$

$$\lfloor n\rfloor=\lfloor n\rfloor\cdot\prod_{\rho}\left(1-\frac{0}{\rho-1}\right)$$

$$D(n)=\lfloor n\rfloor\cdot\prod_{\rho}\left(1-\frac{1}{\rho-1}\right) \tag{4.2}$$

```
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1[n_,z_]:=Sum[z^k/k! P2[n,k],{k,0,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1[n,z]==0,z][[All,2]]
Table[{Sum[MoebiusMu[j],{j,1,n}],n Product[ 1+2/(r-1),{r,zeros[n]}]},{n,4,100}]//TableForm
Table[{1,n Product[ 1+1/(r-1),{r,zeros[n]}]},{n,4,100}]//TableForm
Table[{n,n Product[ 1-0/(r-1),{r,zeros[n]}]},{n,4,100}]//TableForm
Table[{Sum[1,{j,1,n},{k,1,Floor[n/j]}],n Product[ 1-1/(r-1),{r,zeros[n]}]},{n,4,100}]//TableForm
```

A close variant of this idea is

$$\frac{D_z(n)-1}{z}=\sum_{k=1}^{\lfloor \log_2 n\rfloor}\frac{z^{k-1}}{k!}P_{k,2}(n)$$

(4.3)

```
P2[n_,k_]:=Sum[MangoldtLambda[j]/Log[j] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1minus1overz[n_,z_]:=Sum[z^(k-1)/k! P2[n,k],{k,1,Log[2,n]}]
```

This function, for a fixed *n*, has $\log_2 n-1$ roots, denoted ρ. With these roots, $D_z(n)$ is

$$D_z(n)=1+z\cdot\Pi(n)\cdot\prod_{\rho}\left(1-\frac{z}{\rho}\right)$$

(D10)

```
referenced1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
P2[n_,k_]:=Sum[MangoldtLambda[j]/Log[j] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1minus1overz[n_,z_]:=Sum[z^(k-1)/k! P2[n,k],{k,1,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1minus1overz[n,z]==0,z][[All,2]]
D1Alt[n_, z_] := 1 + z P2[n,1] Product[ 1-z/r,{r,zeros[n]}]
Grid[Table[{referenceD1[a=111,s+t I],D1Alt[a,s+t I]},{s,-1.3,4,.7},{t,-1.3,4,.7}]]
```

or

$$D_z(n)=1+z\cdot(\lfloor n\rfloor-1)\cdot\prod_{\rho}\left(1-\frac{z-1}{\rho-1}\right)$$

(D11)

```
referenced1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
P2[n_,k_]:=Sum[MangoldtLambda[j]/Log[j] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1minus1overz[n_,z_]:=Sum[z^(k-1)/k! P2[n,k],{k,1,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1minus1overz[n,z]==0,z][[All,2]]
D1Alt[n_, z_] := 1 + z (n-1) Product[ 1-(z-1)/(r-1),{r,zeros[n]}]
Grid[Table[{referenceD1[a=111,s+t I],D1Alt[a,s+t I]},{s,-1.3,4,.7},{t,-1.3,4,.7}]]
```

and the Riemann Prime counting function is

$$\Pi(n)=(\lfloor n\rfloor-1)\prod_{\rho}\left(1+\frac{1}{\rho-1}\right)$$

(P10)

```
ReferenceRiemannPrimeCnt[n_]:=Sum[MangoldtLambda[j]/Log[j],{j,2,n}]
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1minus1overz[n_,z_]:=Sum[z^(k-1)/k! P2[n,k],{k,1,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1minus1overz[n,z]==0,z][[All,2]]
Table[{N[ReferenceRiemannPrimeCnt[n]],(n-1) Product[ 1+1/(r-1),{r,zeros[n]}]},{n,10,100}]//TableForm
```

More results with these zeros include

$$M(n)=1-(\lfloor n\rfloor-1)\prod_{\rho}(1+\frac{2}{\rho})$$

$$\Pi(n)=(\lfloor n\rfloor-1)\prod_{\rho}(1+\frac{1}{\rho-1})$$

$$\lfloor n\rfloor=1+(\lfloor n\rfloor-1)\prod_{\rho}(1-\frac{0}{\rho-1})$$

$$D(n)=1+2(\lfloor n\rfloor-1)\prod_{\rho}(1-\frac{1}{\rho-1})$$

(4.4)

```
ReferenceRiemannPrimeCnt[n_]:=Sum[MangoldtLambda[j]/Log[j],{j,2,n}]
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1minus1overz[n_,z_]:=Sum[z^(k-1)/k! P2[n,k],{k,1,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1minus1overz[n,z]==0,z][[All,2]]
Table[{Sum[MoebiusMu[j],{j,1,n}],1- (n-1) Product[ 1+2/(r-1),{r,zeros[n]}]},{n,10,100}]//TableForm
Table[{N[ReferenceRiemannPrimeCnt[n]],(n-1) Product[ 1+1/(r-1),{r,zeros[n]}]},{n,10,100}]//TableForm
Table[{n,1+(n-1) Product[ 1-0/(r-1),{r,zeros[n]}]},{n,10,100}]//TableForm
Table[{Sum[1,{j,1,n},{k,1,Floor[n/j]}],1+2(n-1) Product[ 1-1/(r-1),{r,zeros[n]}]},{n,10,100}]//TableForm
```

and

$$M(n)=1-\Pi(n)\cdot\prod_{\rho}(1+\frac{1}{\rho})$$

$$\Pi(n)=\Pi(n)\cdot\prod_{\rho}(1+\frac{0}{\rho})$$

$$\lfloor n\rfloor=1+\Pi(n)\cdot\prod_{\rho}(1-\frac{1}{\rho})$$

$$D(n)=1+2\,\Pi(n)\cdot\prod_{\rho}(1-\frac{2}{\rho})$$

(4.5)

```
ReferenceRiemannPrimeCnt[n_]:=Sum[MangoldtLambda[j]/Log[j],{j,2,n}]
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
D1minus1overz[n_,z_]:=Sum[z^(k-1)/k! P2[n,k],{k,1,Log[2,n]}]
zeros[n_] :=List@@NRoots[ D1minus1overz[n,z]==0,z][[All,2]]
Table[{Sum[MoebiusMu[j],{j,1,n}],1- ReferenceRiemannPrimeCnt[n] Product[ 1+1/r,{r,zeros[n]}]},{n,10,100}]//TableForm
Table[{N[ReferenceRiemannPrimeCnt[n]],ReferenceRiemannPrimeCnt[n] Product[ 1+0/r,{r,zeros[n]}]},{n,10,100}]//TableForm
Table[{n,1+ReferenceRiemannPrimeCnt[n] Product[ 1-1/r,{r,zeros[n]}]},{n,10,100}]//TableForm
Table[{Sum[1,{j,1,n},{k,1,Floor[n/j]}],1+2 ReferenceRiemannPrimeCnt[n] Product[ 1-2/r,{r,zeros[n]}]},{n,10,100}]//TableForm
```

Here's another interesting set of zeros. Starting with $P_{0,2}(n)=1;\ P_{k,2}(n)=\sum_{j=2}^{\lfloor x\rfloor}\frac{\Lambda(j)}{\log j}P_{k-1,2}(\frac{n}{j})$ from (1.6), we define $P_z(n)$, a prime power analog to $D_z(n)$, for which $P_1(n)=\Pi(n)+1$:

$$P_z(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor}\binom{z}{k}P_{k,2}(n)$$

(4.6)

```
bin[ z_, k_ ] := Product[ z-j, { j, 0, k-1 } ]/k!
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
P1[ x_, z_ ] := Sum[ bin[ z, k ] P2[ x, k ], { k, 0, Log[ 2, x ] } ]
```

Although less obvious than the above cases, (4.6) also gives, for fixed *n,* a polynomial of degree $\log_2 n$, using the complex generalized binomial coefficient $\binom{z}{k}=\frac{z(z-1)...(z-k+1)}{k!}$. For example,

$$P_z(12)=\sum_{k=0}^{\lfloor \log_2 12\rfloor}\binom{z}{k}P_{k,2}(12)=(1)+\frac{z}{1!}(\frac{19}{3})+\frac{z\cdot(z-1)}{2!}(8)+\frac{z\cdot(z-1)\cdot(z-2)}{3!}(4)$$

$$P_z(12)=1+\frac{11}{3}z+2z^2+\frac{2}{3}z^3$$

```
bin[ z_, k_ ] := Product[ z-j, { j, 0, k-1 } ]/k!
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
P1[ n_, z_ ] := Sum[ bin[ z, k ] P2[ n, k ], { k, 0, Log[ 2, n ] } ]
Expand[P1[12,z]]
```

Like for $D_z(n)$ above, $P_z(n)$, for fixed *n*, has $\log_2 n$ values of *z* for which $P_z(n)=0$, denoted ρ . With those roots, we have another Riemann Prime counting function identity:

$$\Pi(n)=-1+\prod_{\rho}(1-\frac{1}{\rho})$$

(P11)

```
ReferenceRiemannPrimeCnt[n_]:=Sum[MangoldtLambda[j]/Log[j],{j,2,n}]
bin[ z_, k_ ] := Product[ z-j, { j, 0, k-1 } ]/k!
P2[n_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}];P2[n_,0]:=1
P1[ n_, z_ ] := Sum[ bin[ z, k ] P2[ n, k ], { k, 0, Log[ 2, n ] } ]
zeros[n_] :=List@@NRoots[ P1[n,z]==0,z][[All,2]]
```

```
Table[{N[ReferenceRiemannPrimeCnt[n]],-1+Product[ 1-1/r,{r,zeros[n]}]},{n,4,100}]//TableForm
```

Let's look at one last set of zeros. First, define another sum function, analogous to $D_{k,2}(n)$ but working with logarithms.

$$L_{0,2}(n)=1\,;\ L_{1,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\log j\,;\ L_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}L_{k-1,2}\left(\frac{n}{j}\right) \tag{4.7}$$

```
L2[n_, 0] := 1;L2[n_, 1] := Sum[ Log[j],{j,2,n}]
L2[n_,k_] := Sum[ L2[n/j,k-1],{j,2,n}]
```

Now, define a new function, $L_z(n)$, that relates to $L_{k,2}(n)$ as $D_z(n)$ relates to $D_{k,2}(n)$

$$L_z(n)=\sum_{k=0}^{\lfloor\log_2 n\rfloor}\binom{z}{k}L_{k,2}(n) \tag{4.8}$$

```
L2[n_, 0] := 1;L2[n_, 1] := Sum[ Log[j],{j,2,n}]
L2[n_,k_] := Sum[ L2[n/j,k-1],{j,2,n}]
L1[ n_, z_] := Sum[ Binomial[z,k] L2[n,k],{k,0,Log[2,n]}]
```

Important values of $L_z(n)$ include

$$\log(\lfloor n\rfloor!)=L_1(n)-1\ \text{ and }\ \psi(x)=1-L_{-1}(n) \tag{4.9}$$

where $\psi(n)$ is the second Chebyshev function, $\sum_{j=2}^{\lfloor n\rfloor}\Lambda(j)$.

For some fixed *n*, $L_z(n)=0$ has $\log_2 n$ solutions in *z*, denoted ρ . Using those solutions, we have

$$\log(\lfloor n\rfloor!)=-1+\prod_{\rho}\left(1-\frac{1}{\rho}\right) \tag{4.10}$$

$$\boxed{\psi(n)=1-\prod_{\rho}\left(1+\frac{1}{\rho}\right)} \tag{4.11}$$

```
L2[n_,0]:=1;L2[n_,1]:=Sum[Log[j],{j,2,n}]
L2[n_,k_]:=Sum[L2[n/j,k-1],{j,2,n}]
bin[z_,k_]:=Product[z-j,{j,0,k-1}]/k!
L1[n_,z_]:=Sum[bin[z,k] L2[n,k],{k,0,Log[2,n]}]
zeros[n_]:=List@@NRoots[L1[n,z]==0,z][[All,2]]
```

```
chebyshevReference[n_]:=Sum[MangoldtLambda[j],{j,2,n}]
lognfactReference[n_]:=Sum[Log[j],{j,2,n}]
Table[{Chop[-1+Product[1-1/r,{r,zeros[n]}]],N[lognfactReference[n]],"        ",
Chop[1-Product[1+1/r,{r,zeros[n]}]],N[chebyshevReference[n]]},{n,4,100}]//TableForm
```

5. Computing $\Pi(n)$ with the Hyperbola Method

Here we compute $D_z(n)$ and $\Pi(n)$ pretty quickly, using the Dirichlet Hyperbola Method to $D_{k,2}(n)$. It $D_z(n)$ for any complex z in faster than $O(n)$ time and $O(\log n)$ space. It's especially well suited to computing $\Pi(n)$.

Suppose, given a function $f(n)$, we have this summing function $F_{k,a}(n)$:

$$F_{0,a}(n)=1;\ F_{k,a}(n)=\sum_{j=a}^{\lfloor x\rfloor} f(j)\,F_{k-1,a}(\frac{n}{j}) \tag{5.1}$$

```
Ff[f_, n_, 0, a_] := 1
Ff[f_, n_, k_, a_] := Sum[ f[j] Ff[f, n/j, k-1, a],{j,a,Floor[n]}]
```

$F_{k,a}(n)$ relates to $F_{k,a}(n)$ for successive values of a as

$$F_{k,a}(n)=\sum_{j=0}^{k} \binom{k}{j} f(a)^j F_{k-j,a+1}(\frac{n}{a^j}) \tag{5.2}$$

```
F[f_,n_,0,a_]:=1
F[f_,n_,k_,a_]:=Sum[f[j] F[f,n/j,k-1,a],{j,a,Floor[n]}]
FAlt[f_,n_,k_,a_]:=If[ n < a^k, 0, Sum[Binomial[k,j] f[a]^j F[f,n/a^j,k-j,a+1],{j,0,k}]]
Grid[Table[{F[MoebiusMu, n,k,3],FAlt[MoebiusMu, n,k,3]},{n,10,500,10},{k,1,5}]]
Grid[Table[{F[LiouvilleLambda, n,k,5],FAlt[LiouvilleLambda, n,k,5]},{n,10,500,10},{k,1,5}]]
```

and

$$F_{k,a}(n)=\sum_{j=0}^{k} (-1)^j \binom{k}{j} f(a-1)^j F_{k-j,a-1}(\frac{n}{(a-1)^j}) \tag{5.3}$$

```
F[f_,n_,0,a_]:=1
F[f_,n_,k_,a_]:=Sum[f[j] F[f,n/j,k-1,a],{j,a,Floor[n]}]
FAlt[f_,n_,k_,a_]:=If[ n< a^k, 0, Sum[(-1)^j Binomial[k,j] f[a-1]^j F[f,n/(a-1)^j,k-j,a-1],{j,0,k}]]
Grid[Table[{F[MoebiusMu, n,k,3],FAlt[MoebiusMu, n,k,3]},{n,10,500,10},{k,1,5}]]
Grid[Table[{F[LiouvilleLambda, n,k,5],FAlt[LiouvilleLambda, n,k,5]},{n,10,500,10},{k,1,5}]]
```

A crucial property of (5.1) is

$$F_{k,a}(n)=0 \text{ when } n<a^k$$

(5.4)

```
F[f_,n_,0,a_]:=1
F[f_,n_,k_,a_]:=Sum[f[j] F[f,n/j,k-1,a],{j,a,Floor[n]}]
Grid[Table[ F[MoebiusMu,n,k,2],{n,1,64},{k,1,6}]]
Grid[Table[ F[LiouvilleLambda,n,k,3],{n,1,81},{k,1,6}]]
```

Let's use (5.4) to rewrite (5.2) so the $F_{k,a}(n)$ term on right hand side only uses values of k smaller than the k on the left hand side.

Here's how we do this: let's recursively replace the right hand side reference to $F_{k,a}(n)$ with the identity (5.2) itself, but only when j is 0. Let's keep doing that until $F_{k,a}(n)$ is 0, at which point we'll have the definition we want. It will look like this:

$$\boxed{F_{k,a}(n)=\sum_{j=1}^{k}\binom{k}{j}\sum_{m=a}^{\lfloor n^{\frac{1}{k}}\rfloor} f(m)^j F_{k-j,m+1}\left(\frac{n}{m^j}\right)}$$

(5.5)

```
F[f_,n_,0,a_]:=1
F[f_,n_,k_,a_]:=Sum[f[j] F[f,n/j,k-1,a],{j,a,Floor[n]}]
FAlt[fn_, n_,0,a_]:=1
FAlt[fn_, n_,k_,a_]:=Sum[Binomial[k,j] fn[m]^j FAlt[fn, n/(m^j),k-j,m+1],{j,1,k}, {m,a,Floor[n^(1/k)]}]
id[n_] := 1
Grid[Table[{F[id, n,k, 2],FAlt[id, n,k, 2]},{n,10,500,10},{k,1,7}]]
Grid[Table[{F[MoebiusMu, n,k,3],FAlt[MoebiusMu, n,k, 3]},{n,10,500,10},{k,1,7}]]
```

Should we recursively apply (5.5) to itself until $F_{k,a}(n)$ eliminated entirely (given that $F_{1,a}(n)=1$ and thus $F_{1,a}(n)=\sum_{j=a}^{\lfloor n\rfloor} f(j)$), we'll be left with nested sums of products of $f(n)$ and have, essentially, expressed $F_{k,a}(n)$ with the Dirichlet Hyperbola method.

If $\sum_{j=a}^{\lfloor n\rfloor} f(j)$ can be computed in constant time, this lets us compute $F_{k,a}(n)$ in faster than $O(n)$ time, and in $O(\log n)$ space.

Now, say $f(n)=1$. Apply (5.5) to this function and we have

$$\boxed{\begin{gathered}D_{k,a}(n)=\sum_{j=1}^{k}\binom{k}{j}\sum_{m=a}^{\lfloor n^{\frac{1}{k}}\rfloor} D_{k-j,m+1}\left(\frac{n}{m^j}\right)\\ D_{1,a}(n)=\lfloor n\rfloor-a+1\\ D_{0,a}(n)=1\end{gathered}}$$

(5.6)

```
Dd[n_,0,a_]:=1; Dd[n_,1,a_]:=Floor[n]-a+1
```

```
Dd[n_,k_,a_]:=Sum[Binomial[k,j] Dd[n/(m^(k-j)),j,m+1],{m,a,n^(1/k)},{j,0,k-1}]
ReferenceD1[n_, k_] := Sum[ ReferenceD1[n/j, k-1],{j,1,n}]; ReferenceD1[n_, 0 ] := 1
ReferenceD2[n_, k_] := Sum[ ReferenceD2[n/j, k-1],{j,2,n}]; ReferenceD2[n_, 0 ] := 1
Grid[Table[{Dd[n,k,1],ReferenceD1[n,k]},{n,7,100,5},{k,1,7}]]
Grid[Table[{Dd[n,k,2],ReferenceD2[n,k]},{n,7,100,5},{k,1,7}]]
```

where $D_{k,1}(n)$ is $D_k(n)$ from (1.1) and $D_{k,2}(n)$ is our function from (1.4).

We can use (5.6) with the divisor sum function $D_z(n)$ from (D1) to compute $D_z(n)$ as

$$D_z(n) = \sum_{k=0}^{\lfloor \log_2 n \rfloor} \binom{z}{k} D_{k,2}(n)$$

```
Dd[n_,0,a_]:=1; Dd[n_,1,a_]:=Floor[n]-a+1
Dd[n_,k_,a_]:=Sum[Binomial[k,j] Dd[n/(m^(k-j)),j,m+1],{m,a,n^(1/k)},{j,0,k-1}]
D1[n_,z_]:=Sum[Binomial[z,k] Dd[n,k,2],{k,0,Log[2,n]}]
referenced1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
referenceD1[n_,z_]:=Sum[referenced1[j,z],{j,1,n}]
Grid[Table[{D1[721,s+t I],referenceD1[721,s+t I]},{s,-1.3,4,.7},{t,-1.3,4,.7}]]
```

As Mertens function is $M(n) = D_{-1}(n)$, we can use (5.6) to compute Mertens function as

$$M(n) = \sum_{k=0}^{\lfloor \log_2 n \rfloor} (-1)^k D_{k,2}(n)$$

```
Dd[n_,0,a_]:=1; Dd[n_,1,a_]:=Floor[n]-a+1
Dd[n_,k_,a_]:=Sum[Binomial[k,j] Dd[n/(m^(k-j)),j,m+1],{m,a,n^(1/k)},{j,0,k-1}]
Mertens[n_]:=Sum[(-1)^k Dd[n,k,2],{k,0,Log[2,n]}]
MertensReference[n_]:=Sum[MoebiusMu[j],{j,1,n}]
Grid[Table[{Mertens[n],MertensReference[n]},{n,2,100}]]
```

and we can use (5.6) with (P3) to compute Riemann's Prime counting function as

$$\Pi(n) = \sum_{k=1}^{\lfloor \log_2 n \rfloor} \frac{(-1)^{k+1}}{k} D_{k,2}(n)$$

```
Dd[n_,0,a_]:=1;Dd[n_,1,a_]:=Floor[n]-a+1
Dd[n_,k_,a_]:=Sum[Binomial[k,j] Dd[n/(m^(k-j)),j,m+1],{m,a,n^(1/k)},{j,0,k-1}]
RiemannPrimeCount[n_]:=Sum[(-1)^(k-1)/k Dd[n,k,2],{k,1,Log[2,n]}]
ReferenceRiemannPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
```

```
Grid[Table[{RiemannPrimeCount[n],ReferenceRiemannPrimeCount[n]},{n,2,100}]]
```

and we can use (3.1) to compute $P_{0,2}(n)=1;\ P_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}\frac{\Lambda(j)}{\log j}P_{k-1,2}(\frac{n}{j})$ from (1.6) as

$$P_{j,2}(n)=\sum_{k=0}^{\lfloor \log_2 n\rfloor}\left(\frac{\partial^k}{\partial y^k}(\log(1+y))^j \text{ at } y=0\right)\cdot D_{k,2}(n) \tag{5.7}$$

```
P2[n_,0]:=1
P2[n_,k_]:=P2[n,k]=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] P2[n/j,k-1],{j,2,Floor[n]}]
Dd[n_,0,a_]:=1;Dd[n_,1,a_]:=Floor[n]-a+1
Dd[n_,k_,a_]:=Sum[Binomial[k,j] Dd[n/(m^(k-j)),j,m+1],{m,a,n^(1/k)},{j,0,k-1}]
P2Alt[n_,j_]:=Sum[1/k!(D[Log[1+x]^j,{x,k}]/.x®0) Dd[n,k,2],{k,0,Log[2,n]}]
Grid[Table[{P2[n,k],P2Alt[n,k]},{n,10,500, 10},{k,1,5}]]
```

(5.6) lets us compute $D_z(n)$, $M(n)$, and $P_{k,2}(n)$ in faster than $O(n)$ time, and in $O(\log n)$ space. In fact, once we've computed our $\log_2 n$ values of $D_{k,2}(n)$ in $O(n)$ time, we can compute *any* value of $D_z(n)$ or $P_{k,2}(n)$ in $O(\log_2 n)$ operations, via the identities above.

This idea works especially well at computing $\Pi(n)$ if we apply a wheel to (5.6). If the sum in (5.6) is only taken over m=numbers not divisible by, say, the first 8 primes, and the same wheel is applied to $D_{1,a}(n)$ in (5.6), the algorithm runs, empirically, in something like $O(n^{\frac{4}{5}})$ time, and speeds up between x1000 and x10000 in constant time terms.

A pretty fast C implementation using this idea to count primes is at http://icecreambreakfast.com/primes/NMPrimeCounter.cpp

Other descriptions of this technique are in section E.2 of http://www.icecreambreakfast.com/primecount/PrimeCountingSurvey.pdf , and section 4-2 of http://www.icecreambreakfast.com/primecount/LinnikVariations.pdf

6. Computing $\Pi(n)$ with Another Combinatorial Approach

Taking inspiration from an algorithm published by Marc Deléglise and Joöl Rivat in their paper "Computing the summation of the Mobius function" (see http://projecteuclid.org/download/pdf_1/euclid.em/1047565447) this section details a combinatorial method for computing $D_z(n)$ and $\Pi(n)$ in $O(n^{2/3}\log n)$ time and $O(n^{1/3}\log n)$ space. That paper makes a handy reference for what follows. This section is particularly hard to follow, owing to its complexity.

Suppose, for some function $f(n)$, we have the following summatory functions

$$F_0(n)=1\,;\; F_k(n)=\sum_{j=1}^{\lfloor n\rfloor} f(j)\,F_{k-1}\left(\frac{n}{j}\right) \qquad f_k(n)=F_k(n)-F_k(n-1)$$

(6.1)

```
F[ fn_, n_, 0] := 1; F[ fn_, n_, k_] := Sum[ fn[j] F[fn, n/j, k-1],{j,1,Floor[n]}]
f[ fn_, n_, k_] := F[ fn,x,k]-F[fn,n-1,k]
```

Then the following rather complicated combinatorial identity holds for $F_k(n)$, with $1<t<n$:

$$F_k(n)=F_k(t)+\sum_{j=t+1}^{\lfloor n\rfloor} f(j)F_{k-1}\left(\frac{n}{j}\right)+\sum_{j=1}^{t}\sum_{s=\lfloor \frac{t}{j}\rfloor+1}^{\lfloor \frac{n}{j}\rfloor}\sum_{m=1}^{k-1} f(s)\,f_m(j)\,F_{k-m-1}\left(\frac{n}{js}\right)$$

(6.2)

```
F[fn_,n_,0]:=1;F[fn_,n_,k_]:=F[fn,n,k]=Sum[fn[j] F[fn,n/j,k-1],{j,1,Floor[n]}]
f[fn_,n_,k_]:=F[fn,n,k]-F[fn,n-1,k]
FAlt[fn_,n_,k_,t_]:=F[fn, t,k]+Sum[fn[j] F[fn,n/j,k-1],{j,t+1,Floor[n]}]+Sum[fn[s]f[fn,j,m]F[fn,n/(j s),k-m-1],{j,1,t},{s,Floor[t/j]
+1,Floor[n/j]},{m,1,k-1}]
id[n_] := 1
Grid[Table[{F[id, n,k],FAlt[id, n,k, Floor[ n^(1/2)]]},{n,10,500,10},{k,1,7}]]
Grid[Table[{F[MoebiusMu, n,k],FAlt[MoebiusMu, n,k, Floor[ n^(1/2)]]},{n,10,500,10},{k,1,7}]]
```

Inspection shows that, in (6.2), the largest argument for $F_k(n)$ is $\frac{n}{t}$, and the largest for $f_k(n)$ is t. The only exceptions are for $F_1(n)$, where the largest argument is *n,* and $f(n)$, which also takes arguments up to *n*.

The reason this identity is useful is that, if $f(n)$ and $F_1(n)=\sum_{j=1}^{n} f(j)$ can be computed in constant time, and if we have some external method to compute a table of $F_k(n)$ up to arguments of $\frac{n}{t}$ and $f_k(n)$ up to arguments of t , then we can use (6.2) to compute $F_k(n)$.

Our goal is to compute $F_k(n)$ with (6.2) as efficiently as possible, so we need another important identity. Inspection of $F_k(n)$ in (6.1) should make clear that $F_k(n)=F_k(\lfloor n\rfloor)$. Now, it's the case that if we have functions $g(n)$ and $h(n)$ such that $g(n)=g(\lfloor n\rfloor)$ and $h(n)=h(\lfloor n\rfloor)$, then the sum $\sum_{j=1}^{\lfloor n\rfloor}(g(j)-g(j-1))h(\frac{n}{j})$ can be split into two parts as

$$\sum_{j=1}^{\lfloor n\rfloor}(g(j)-g(j-1))h(\frac{n}{j})=\sum_{j=1}^{\lfloor n^{\frac{1}{2}}\rfloor}(g(j)-g(j-1))h(\frac{n}{j})+\sum_{j=1}^{\lfloor \frac{n}{\lfloor n^{\frac{1}{2}}\rfloor}\rfloor-1}(g(\frac{n}{j})-g(\frac{n}{j+1}))\cdot h(j) \tag{6.3}$$

```
s1[ n_, g_, h_ ] := Sum[ (g[j]-g[j-1]) h[ n/j],{j,1,Floor[n]}]
s2[ n_, g_, h_ ] := Sum[ (g[j]-g[j-1]) h[ n/j], {j, 1, Floor[ n^(1/2)]}]+Sum[ (g[n/j]-g[n/(j+1)])h[j], {j,1,Floor[ n/Floor[( n^(1/2))]]-1}]
id[ n_ ] := Floor[n]
mert[n_] := Sum[ MoebiusMu[ j ], {j,1,Floor[n]}]
Table[ { s1[ n, id, id ], "=", s2[n, id, id], "  ", s1[ n,mert,mert ], "=", s2[n, mert,mert] }, {n,100,1000,100}]// TableForm
```

Variants of (6.3) can be applied to two of the sums in (6.2), as long as t is less than $n^{\frac{1}{2}}$, to leave it as

$$\begin{aligned}F_k(n)=F_k(t)+&\\ &\sum_{j=t+1}^{\lfloor n^{\frac{1}{2}}\rfloor} f(j)\cdot F_{k-1}(\frac{n}{j})+\sum_{j=1}^{\lfloor \frac{n}{\lfloor x^{\frac{1}{2}}\rfloor}-1\rfloor}(F_1(\frac{n}{j})-F_1(\frac{n}{j+1}))F_{k-1}(j)\\ &+\sum_{j=1}^{t}\sum_{s=\lfloor \frac{t}{j}+1\rfloor}^{\lfloor\lfloor \frac{n}{j}\rfloor^{\frac{1}{2}}\rfloor}\sum_{m=1}^{k-1} f(s)\cdot f_m(j)\cdot F_{k-m-1}(\frac{n}{js})+\sum_{j=1}^{t}\sum_{s=1}^{\lfloor \frac{\lfloor \frac{n}{j}\rfloor}{\lfloor\lfloor \frac{n}{j}\rfloor^{\frac{1}{2}}\rfloor}-1\rfloor}(F_1(\frac{n}{js})-F_1(\frac{n}{j(s+1)}))\cdot\sum_{m=1}^{k-1} f(s)\cdot f_m(j)\cdot F_{k-m-1}(s)\end{aligned} \tag{6.4}$$

```
F[fn_,n_,k_, s_]:=F[fn,n,k,s]=Sum[(fn[m]^(k-j)) Binomial[k,j]F[ fn,n/(m^(k-j)),j, m+1],{m,s,n^(1/k)},{j,0,k-1}]
F[fn_,n_,0, s_]:=1
F[fn_,n_,k_]:= F[fn,n,k,1]
f[fn_,n_, k_]:=F[fn,n,k]-F[fn,n-1,k]
FAlt[fn_, n_, k_, t_]:=F[fn,t, k]+Sum[fn[j] F[fn, n/j, k-1],{j,t+1,n^(1/2)}]+
Sum[Sum[fn[m],{m,Floor[n/(j+1)]+1,n/j}]F[fn, j, k-1],{j,1,n/Floor[n^(1/2)]-1}]+
 Sum[fn[s] f[fn, j, m] F[fn,n/(j s),k-m-1],{j,1,t},{s,Floor[t/j]+1,Floor[n/j]^(1/2)},{m,1,k-1}]+Sum[(Sum[fn[m],{m,Floor[n/(j(s+1))]+1,n/(j
s)}])(Sum[f[fn,j, m] F[fn,s, k-m-1],{m,1,k-1}]),{j,1,t},{s,1,Floor[n/j]/Floor[Floor[n/j]^(1/2)]-1}]
```

```
FAlt[fn_, n_, 1, t_]:=Sum[fn[j],{j,1,n}]
Grid[Table[{F[MoebiusMu, n,k,1],FAlt[MoebiusMu, n,k,Floor[n^(1/3)]]},{n,10,500,10},{k,1,7}]]
Grid[Table[{F[LiouvilleLambda, n,k,1],FAlt[LiouvilleLambda, n,k,Floor[n^(1/3)]]},{n,10,500,10},{k,1,7}]]
```

Now let's define our function $f(n)$ and choose a value for *t*.

Our value for *t* will be $n^{\frac{1}{3}}$.

Our function $f(n)$ will be $f(n)=0\,if\; n=1, 1\,otherwise$. This will satisfy our requirement that $f(n)$ be computable for any value of *n* in constant time.

Thus, our function $F_1(n)$ will be $F_1(n)=\lfloor n\rfloor-1$, also computable for any *n* in constant time.

Our function $F_k(n)$ will be $D_{k,2}(n)$, defined in (1.4).

And our function $f_k(n)=F_k(n)-F_k(n-1)$, which is just $D_{k,2}(n)-D_{k,2}(n-1)$ can also be defined as

$$d_{k,2}(n)=\sum_{j|n} d_{k-1,2}(j)\,d_{1,2}\left(\frac{n}{j}\right) \qquad d_{1,2}(n)=1\,if\; n>1, 0\,otherwise \qquad d_{0,2}(n)=1\,if\; n=1, 0\,otherwise \tag{6.5}$$

```
d2[ n_, k_ ] := Sum[ d2[ j, k-1 ] d2[ n/j, 1 ], { j, Divisors[ n ] } ];d2[ n_, 1 ] := If[ n>1, 1,0];d2[ n_, 0 ] := 0;d2[ 1, 0 ] := 1
Grid[ Table[ d2[ n, k ], { n, 1, 50 }, { k, 1, 7 } ] ]
```

Applying this all to (6.4), we have

$$D_{k,2}(n)=D_{k,2}(t)+ \sum_{j=\lfloor n^{\frac{1}{3}}\rfloor+1}^{\lfloor n^{\frac{1}{2}}\rfloor} D_{k-1,2}\left(\frac{n}{j}\right) + \sum_{j=1}^{\lfloor \frac{n}{\lfloor n^{\frac{1}{2}}\rfloor}-1\rfloor} \left(\lfloor\frac{n}{j}\rfloor-\lfloor\frac{n}{j+1}\rfloor\right) D_{k-1,2}(j)$$
$$+\sum_{j=2}^{\lfloor n^{\frac{1}{3}}\rfloor} \sum_{s=\lfloor \frac{\lfloor n^{\frac{1}{3}}\rfloor}{j}+1\rfloor}^{\lfloor\lfloor \frac{n}{j}\rfloor^{\frac{1}{2}}\rfloor} \sum_{m=1}^{k-1} d_{m,2}(j)\cdot D_{k-m-1,2}\left(\frac{n}{js}\right) +\sum_{j=2}^{\lfloor n^{\frac{1}{3}}\rfloor} \sum_{s=1}^{\lfloor \frac{\lfloor \frac{n}{j}\rfloor}{\lfloor\lfloor \frac{n}{j}\rfloor^{\frac{1}{2}}\rfloor}-1\rfloor} \left(\lfloor\frac{n}{js}\rfloor-\lfloor\frac{n}{j(s+1)}\rfloor\right)\cdot \sum_{m=1}^{k-1} d_{m,2}(j)\cdot D_{k-m-1,2}(s) \tag{6.6}$$

```
D2[n_,k_]:=D2[n,k]=Sum[D2[n/j,k-1],{j,2,Floor[n]}];D2[n_,0]:=1
d2[n_,k_]:=D2[n,k]-D2[n-1,k]
D2Alt[n_,k_]:=D2[ n^(1/3), k]+Sum[D2[n/j,k-1],{j,Floor[n^(1/3)]+1,n^(1/2)}]+Sum[(Floor[n/j]-Floor[n/(j+1)])D2[j,k-1],
{j,1,n/Floor[n^(1/2)]-1}]+Sum[d2[j,m] D2[n/(j s),k-m-1],{j,2,n^(1/3)},{s,Floor[Floor[n^(1/3)]/j]+1,Floor[n/j]^(1/2)},{m,1,k-1}]
+Sum[(Floor[n/(j s)]-Floor[n/(j(s+1))])(Sum[d2[j,m] D2[s,k-m-1],{m,1,k-1}]),{j,2,n^(1/3)},{s,1,Floor[n/j]/Floor[Floor[n/j]^(1/2)]-1}]
D2Alt[n_,1]:=Floor[n]-1
```

```
Grid[Table[{D2[n,k],D2Alt[n,k]},{n,10,500,10},{k,1,7}]]
```

It is hopefully not too much of a stretch to suggest that if we already had a table with all values for $d_{j,2}(n)$ up to arguments of $n^{\frac{1}{3}}$, for $2 \le j \le k$, and if we had already had a table with all values of $D_{j,2}(n)$ up to arguments of $n^{\frac{2}{3}}$, for $2 \le j \le k$, and taking into account that $D_{k,2}(n)=0$ when $n<2^k$, that (6.6) should be able to compute $D_{k,2}(n)$ for any k in something like $O(n^{\frac{2}{3}} \log n)$ time complexity.

So how would we compute such a table, with values of $d_{k,2}(n)$ up to arguments of $n^{\frac{1}{3}}$, and values of $D_{k,2}(n)$ up to arguments of $n^{\frac{2}{3}}$?

Well, suppose we had a number in prime factored form, $n=\prod_{p^\alpha|n} p^\alpha$. Then we can express $d_k(n)$, the function from (1.3), as

$$d_k(n)=\prod_{p^\alpha|n} (-1)^\alpha \binom{-k}{\alpha} \tag{6.7}$$

```
d1[ n_, z_ ] := Product[ (-1)^p[ [ 2 ] ] Binomial[ -z, p[ [ 2 ] ] ], { p, FI[ n ] } ];FI[ n_ ] := FactorInteger[ n ];FI[ 1 ] := { }
Grid[Table[d1[n,k],{n,1,50},{k,1,7}]]
```

and $d_{k,2}(n)$, from (6.5), in terms of $d_k(n)$ as

$$d_{k,2}(n)=\sum_{j=0}^{k} (-1)^j \binom{k}{j} d_{k-j}(n) \tag{6.8}$$

```
d2[n_,k_]:=Sum[d2[j,k-1] d2[n/j,1],{j,Divisors[n]}];d2[n_,1]:=If[n>1,1,0];d2[n_,0]:=0;d2[1,0]:=1
d1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
d2Alt[n_, k_] := Sum[ (-1)^j Binomial[k,j] d1[ n, k-j], {j,0, k}]
Grid[Table[ {d2[n,k], d2Alt[n,k]},{n,1,50},{k,1,7}]]
```

and using $d_{k,2}(n)$, we can express $D_{k,2}(n)$ as

$$D_{k,2}(n)=D_{k,2}(n-1)+d_{k,2}(n) \tag{6.9}$$

```
D2[ n_, k_ ] := Sum[ D2[ n/j, k-1 ], { j, 2, Floor[ n ] } ];D2[ n_, 0 ] := 1
d1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
d2[n_, k_] := Sum[ (-1)^j Binomial[k,j] d1[ n, k-j], {j,0, k}]
```

```
D2Alt[ n_, k_ ] := D2[ n-1, k ] + d2[ n, k ]
Grid[Table[ {D2[n,k], D2Alt[n,k]},{n,1,50},{k,1,7}]]
```

So, if we had some way to get numbers in prime factored form and then applied that process sequentially from 1 to $n^{\frac{2}{3}}$, we could use (6.6), (6.7), and (6.8) to build a table of values of $D_{k,2}(n)$ up to arguments of $n^{\frac{2}{3}}$.

And in fact, we can use a suitable variant of the Sieve of Eratosthenes to do just that.

All told, with sieving and the above three identities, we can compute $D_{j,2}(n)$ for $2 \le j \le k$ for arguments from 1 to $n^{\frac{2}{3}}$ in something like $O(n^{2/3} \log n)$ time and $O(n^{2/3} \log n)$ space.

We can improve our performance bound to $O(n^{1/3} \log n)$ space if we use a segmented sieve and re-order the way that (6.6) is calculated so that values of $D_{k,2}(n)$ are applied from smallest arguments to largest, with that application interleaved with sieving of blocks of size $n^{\frac{1}{3}}$.

The identity (6.6), coupled with sieving, computes $D_{k,2}(n)$. We can then use our identity (D1),

$$D_z(n) = \sum_{k=0}^{\lfloor \log_2 n \rfloor} \binom{z}{k} D_{k,2}(n)$$

to compute the generalized divisor function $D_z(n)$ for any *z*, or our identity (P3) to compute the Riemann Prime counting function as

$$\Pi(n) = \sum_{k=1}^{\lfloor \log_2 n \rfloor} \frac{(-1)^{k+1}}{k} D_{k,2}(n)$$

Computing $D_{j,2}(n)$ for $2 \le j \le k$ at once, in bulk, presents opportunities for caching and simplification. Thus, the above process can be used to compute $D_z(n)$ for any *z*, as well as $\Pi(n)$, in something like $O(n^{2/3} \log n)$ time and $O(n^{1/3} \log n)$ space.

A C implementation of this algorithm, being used to count primes in the advertised time and space bounds of $O(n^{2/3} \log n)$ time and $O(n^{1/3} \log n)$ space can be found at http://www.icecreambreakfast.com/primecount/primescode.html . Further descriptions of this technique, with better justifications of the combinatorial identities, can be found in http://www.icecreambreakfast.com/primecount/PrimeCounting_NathanMcKenzie.pdf

7. Expressing the Difference between $li(n)$ and $\Pi(n)$: Smoothing

This section shows one way to express the exact difference between $\Pi(n)$ and the logarithmic integral $li(n)$, through a certain notion of smoothing from discrete integers to smooth real-valued continuum. Because operations on zeta functions are generally more familiar, we'll work through the ideas in the section using $\zeta(s)$ first as a road map, then use those same techniques on $D_z(n)$. The section then concludes with a similar identity connecting the Chebyshev function $\psi(n)$ and n.

Here's our plan of attack for the following section. We begin with $\zeta(s)-1$, show an approach for smoothing it, and then express $\zeta(s)-1$ as two terms, its smoothed part and then the rest. Then we do the same for $(\zeta(s)-1)^k$. We then use that to express identities for $\zeta(s)^k$ and $\log\zeta(s)$ that also have a smoothed part and then the rest.

This won't lead to novel results about $\zeta(s)^k$ or $\log\zeta(s)$, but it will map out the approach we'll then use to show $li(n)$ as the smoothed part of $\Pi(n)$. In what follows, we'll ignore questions of convergence.

So let's begin with the Riemann Zeta function.
First, we start with $\zeta(s)-1$.

$$\zeta(s)-1=\frac{1}{2^s}+\frac{1}{3^s}+\frac{1}{4^s}+\ldots \tag{7.1}$$

`{Zeta[s]-1,Sum[ j^-s,{j,2,Infinity}]}`

We can think of $\zeta(s)-1$ as taking the function $\frac{1}{x^s}$ and sampling and discretizing it at regular intervals of width 1, where the first sample takes its value at 1 plus our sample width.

To smooth our function, we sample more frequently, say with a width of ½, keeping the constraint that the first sample is taken at 1 plus our sample width. So our new, smoothed sum is

$$\frac{1}{2}\cdot\left(\frac{1}{(\frac{3}{2})^s}+\frac{1}{(\frac{4}{2})^s}+\frac{1}{(\frac{5}{2})^s}+\frac{1}{(\frac{6}{2})^s}+\ldots\right) \tag{7.2}$$

`(1/2) Sum[ 1/((k+3)/2)^s, {k,0,Infinity}]`

A bit of algebra reveals this is

$$\frac{1}{2}\cdot\left(\frac{2^s}{3^s}+\frac{2^s}{4^s}+\frac{2^s}{5^s}+\frac{2^s}{6^s}+\ldots\right)=\frac{2^s}{2}\cdot\left(\frac{1}{3^s}+\frac{1}{4^s}+\frac{1}{5^s}+\frac{1}{6^s}+\ldots\right)=2^{s-1}\zeta(s,3)$$

(7.3)

```
{(1/2) Sum[ 1/((k+3)/2)^s, {k,0,Infinity}],2^(-1+s) Zeta[s,3]}
```

where $\zeta(s,a)$ is the Hurwitz Zeta function, defined as $\zeta(s,y)=\sum_{n=0}^{\infty}\frac{1}{(n+y)^s}$. The approach in (7.3) generalizes to

$$\frac{1}{y}\left(\frac{1}{\left(\frac{y+1}{y}\right)^s}+\frac{1}{\left(\frac{y+2}{y}\right)^s}+\frac{1}{\left(\frac{y+3}{y}\right)^s}+\frac{1}{\left(\frac{y+4}{y}\right)^s}+\ldots\right)=y^{s-1}\zeta(s,y+1)$$

(7.4)

```
Table[((1/y) Sum[ 1/((y+k)/y)^s, {k,1,Infinity}])-(y^(-1+s) Zeta[s,y+1]),{y,1,6,1/3}]//TableForm
```

If we name the right hand side of (7.4) as

$$c(s,y)=x^{s-1}\zeta(s,y+1)$$

(7.5)

```
c[s_,y_]:=y^(s-1) HurwitzZeta[s,y+1]
```

then obviously

$$c(s,1)=\zeta(s)-1$$

```
c[s_,y_]:=y^(s-1) HurwitzZeta[s,y+1]
{c[s,1], Zeta[s]-1}
```

and, as *x* gets larger,

$$\lim_{y\to\infty} c(s,y)=\int_1^{\infty}\frac{1}{y^s}dy=\frac{1}{s-1}$$

(7.6)

```
c[s_,y_]:=y^(s-1) HurwitzZeta[s,y+1]
{Limit[ c[s,y],y->Infinity],1/(s-1)}
```

Thus, we can express $\zeta(s)-1$ as $c(s,1)=\left(\lim_{y\to\infty} c(s,y)\right)-\int_1^{\infty}\frac{\partial}{\partial y}c(s,y)dy$, which is to say,

$$\zeta(s)-1=\frac{1}{s-1}-\int_1^{\infty}\frac{\partial}{\partial y}\left(y^{s-1}\zeta(s,y+1)\right)dy$$

(7.7)

```
Table[{Zeta[s]-1, 1/(s-1) - Integrate[ D[ y^(s-1) Zeta[ s, y+1],y],{y,1,Infinity}]},{s,2,6}]
```

Let's take the same approach with $(\zeta(s)-1)^z$. Starting with (7.5), we have

$$c(s,y)^z = y^{z(s-1)}\zeta(s,y+1)^z \tag{7.8}$$

```
c[s_,y_]^z:=y^z(s-1) HurwitzZeta[s,y+1]^z
```

Now, it must be that

$$c(s,1)^z = (\zeta(s)-1)^z$$

```
c[s_,x_]:=y^(s-1) HurwitzZeta[s,y+1]
{c[s,1]^z, (Zeta[s]-1)^z}
```

Somewhat more work shows that, as *x* gets larger and the samples get smaller, we have

$$\lim_{y\to\infty} c(s,y)^2 = \int_1^\infty\int_1^\infty \frac{1}{(x\,z)^s}\,dz\,dx = \frac{1}{(s-1)^2}$$

$$\lim_{y\to\infty} c(s,y)^3 = \int_1^\infty\int_1^\infty\int_1^\infty \frac{1}{(x\,z\,w)^s}\,dw\,dz\,dx = \frac{1}{(s-1)^3}$$

and, more generally,

$$\lim_{y\to\infty} c(s,y)^z = \frac{1}{(s-1)^z} \tag{7.9}$$

```
c[s_,y_]:=y^(s-1) HurwitzZeta[s,y+1]
{Limit[ c[s,y]^z,y->Infinity],1/(s-1)^z}
```

Thus, following the pattern of (7.7), our identity for $(\zeta(s)-1)^z$ is

$$(\zeta(s)-1)^z = \frac{1}{(s-1)^z} - \int_1^\infty \frac{\partial}{\partial y}\left(y^{z(s-1)}\zeta(s,y+1)^z\right)dy \tag{7.10}$$

```
f[s_, z_]:=N[1/((s-1)^z)-Integrate[ D[ y^(z(s-1))Zeta[s,y+1]^z,y],{y,1,Infinity}]]
g[s_, z_] := N[(Zeta[s]-1)^z]
Grid[Table[{f[s,z],g[s,z]},{s,2,4},{z,1,4}]]
```

Now let's address $\zeta(s)^z$ and $\log\zeta(s)$. The generalized binomial theorem gives

$$\zeta(s)^z = \sum_{k=0}^{\infty} \binom{z}{k} (\zeta(s)-1)^k \tag{7.11}$$

```
Grid[Table[{Zeta[s]^(i + j I),Sum[ Binomial[ i + j I, k]( Zeta[s]-1)^k, {k,0,Infinity}]},{i,2,4},{j,1,4}]]
```

Making use of (7.10) and (7.11) means $\zeta(s)^z$ is

$$\zeta(s)^z = \sum_{k=0}^{\infty} \binom{z}{k} \left(\frac{1}{(s-1)^k} - \int_1^{\infty} \frac{\partial}{\partial y} (y^{k(s-1)} \zeta(s, y+1)^k) \, dy \right) \tag{7.12}$$

which, with a bit of work, simplifies to

$$\zeta(s)^z = \left(\frac{s}{s-1} \right)^z - \sum_{k=0}^{\infty} \int_1^{\infty} \binom{z}{k} \frac{\partial}{\partial y} (y^{k(s-1)} \zeta(s, y+1)^k) \, dy \tag{7.13}$$

Because $\log \zeta(s) = \lim_{z \to 0} \frac{\zeta(s)^z - 1}{z}$, with more work, we can use (7.12) for the following identity for $\log \zeta(s)$:

$$\log \zeta(s) = \log\left(\frac{s}{s-1} \right) - \sum_{k=1}^{\infty} \int_1^{\infty} \frac{(-1)^{k-1}}{k} \frac{\partial}{\partial y} (y^{k(s-1)} \zeta(s, y+1)^k) \, dy \tag{7.14}$$

Now let's start again, this time showing how the above process maps to $D_{k,2}(n)$, $D_z(n)$ and $\Pi(n)$.

Start with $D_{1,2}(x) = \sum_{j=2}^{\lfloor x \rfloor} 1 = \sum_{j=2}^{\lfloor x \rfloor} j^0 = 1 \cdot ((2)^0 + (3)^0 + (4)^0 + (5)^0 + \ldots + (x)^0)$ (7.15)

which maps to $\zeta(s)-1$. Let's go through the same process we did in (7.1) through (7.4), taking more and smaller samples to smooth this sum, with the constraint that the first sample is taken at 1 plus the width of our sample.

So, taking (7.15) but smoothing it by using samples of width ½, we'd have

$$\frac{1}{2} \cdot \left(\left(\frac{3}{2}\right)^0 + \left(\frac{4}{2}\right)^0 + \left(\frac{5}{2}\right)^0 + \left(\frac{6}{2}\right)^0 + \ldots + \left(\frac{2\mathrm{n}}{2}\right)^0 \right) = \frac{1}{2} \sum_{j=3}^{\lfloor 2\mathrm{n} \rfloor} 1 = \frac{1}{2} D_{1,3}(2n) \tag{7.16}$$

```
D3[n_,k_]:=Sum[D2[n/j,k-1],{j,3,Floor[n]}];D3[n_,0]:=1
{ (1/2)D3[ 2n, 1], (1/2)Sum[ 1,{j,3,Floor[2n]}]}
```

Now, (7.15) can be rewritten, if we'd like, as $D_{1,2}(n)=\sum_{j=0}^{\lfloor n-2\rfloor}(j+2)^0$. Using this form, we can generalize from *2* to some real valued *y*, giving us

$$D_{1,y}(x)=\sum_{j=0}^{\lfloor x-y\rfloor}(j+y)^0 \tag{7.17}$$

```
Dd[x_, 1, y_] := Sum[ (j+y)^0, {j,0, Floor[x-y]}]
```

$D_{1,a}(x)$ plays a role here that, previously, $\zeta(s,a)$, the Hurwitz Zeta function from (7.3), did. Using it just as we did in (7.4), we can generalize(7.16) to

$$\frac{1}{y}\cdot\left(\left(\frac{y+1}{y}\right)^0+\left(\frac{y+2}{y}\right)^0+\left(\frac{y+3}{y}\right)^0+\left(\frac{y+4}{y}\right)^0+\ldots+\left(\frac{y\,x}{y}\right)^0\right)=\frac{1}{y}\sum_{j=y+1}^{\lfloor y\cdot x\rfloor}1=y^{-1}D_{1,y+1}(x\cdot y) \tag{7.18}$$

```
Dd[x_,1,y_]:=Sum[(j+y)^0,{j,0,Floor[x-y]}]
Grid[Table[ { y^-1 Sum[ 1,{j,y+1,Floor[y x]}], y^-1 (Dd[x y, 1, y+1])},{x,1,40},{y,1,5}]]
```

Following the steps in (7.5) through (7.7), if we name the right hand side of (7.18) as

$$C_{1,y}(x)=y^{-1}D_{1,y+1}(x\cdot y) \tag{7.19}$$

```
Dd[x_, 1, y_] := Sum[ (j+y)^0, {j,0, Floor[x-y]}]
Cc[ x_, 1, y_] := y^(-1) Dd[ x y, 1, y+1]
```

we see that

$$C_{1,1}(x)=D_{1,2}(x)=\lfloor x\rfloor-1 \tag{7.20}$$

```
Dd[x_,1,y_]:=Sum[(j+y)^0,{j,0,Floor[x-y]}]
Cc[x_,1,y_]:=y^(-1) Dd[x y,1,y+1]
{Floor[x]-1,Cc[x,1,1]}
```

and

$$\lim_{x\to\infty}C_{1,y}(x)=\int_1^\infty dy=x-1 \tag{7.21}$$

```
Dd[x_,1,y_]:=Sum[(j+y)^0,{j,0,Floor[x-y]}]
Cc[x_,1,y_]:=y^(-1) Dd[x y,1,y+1]
Table[ {n/7-1, Limit[ Cc[n/7,1,z], z->Infinity]}, {n,1,20}]//TableForm
```

which means

$$D_{1,2}(x)=x-1-\int_1^\infty\frac{\partial}{\partial y}\left(y^{-1}D_{1,y+1}(x\,y)\right)dy$$

(7.22)

which is not a particularly interesting result.

Our next step, in the zeta version, was to go from $\zeta(s)-1$ to $(\zeta(s)-1)^k$. In our new, convolution context, that means going from smoothing $D_{1,2}(x)$ to smoothing $D_{k,2}(x)$.

For that, we need a Dirichlet convolution equivalent to $(\zeta(s,a))^k$, the Hurwitz Zeta function raised to positive integer powers as in (7.8).

Our equivalence will be the following function, taken at real values for *a*, which (7.17) was a special case of:

$$D_{0,y}(x)=1\,;\; D_{k,y}(x)=\sum_{j=0}^{\lfloor x-y\rfloor} D_{k-1,y}\left(\frac{x}{j+y}\right)$$

(7.23)

```
Dd[x_, 0, y_] := 1
Dd[x_, k_, y_] := Sum[ Dd[x/(j+y),k-1,y],{j,0,Floor[x-y]}]
```

For the sake of fast computation, if *y* is a positive integer, we can compute $D_{k,y}(x)$ with our identity from (5.6),

$$D_{k,y}(x)=\sum_{j=1}^{k}\binom{k}{j}\sum_{m=y}^{\lfloor x^{\frac{1}{k}}\rfloor} D_{k-j,m+1}\left(\frac{x}{m^j}\right)$$
$$D_{1,y}(x)=\lfloor x\rfloor-y+1$$
$$D_{0,y}(x)=1$$

```
Dd[x_, 0, y_] := 1
Dd[x_, k_, y_] := Sum[ Dd[x/(j+y),k-1,y],{j,0,Floor[x-y]}]
DdAlt[x_,0,y_]:=1; DdAlt[x_,1,y_]:=Floor[x]-y+1
DdAlt[x_,k_,y_]:=DdAlt[x,k,y]=Sum[Binomial[k,j] DdAlt[x/(m^(k-j)),j,m+1],{m,y,x^(1/k)},{j,0,k-1}]
Grid[Table[{Dd[x,k,1],DdAlt[x,k,1]},{x,7,100,5},{k,1,7}]]
```

With identity (723), we can write our smoothing function, which generalizes (7.19) and is analogous to (7.8), as

$$C_{k,y}(x)=y^{-k}D_{k,y+1}(x\,y^k)$$

(7.24)

```
Dd[x_, 0, y_] := 1
Dd[x_, k_, y_] := Sum[ Dd[x/(j+y),k-1,y],{j,0,Floor[x-y]}]
Cc[x_, k_, y_] := y^-k Dd[ x y^k, k, y+1]
```

where *y* determines, essentially, how smoothed the function is. If $y=1$, meaning entirely unsmoothed, we have

$$\boxed{C_{k,1}(x)=D_{k,2}(x)}$$

(7.25)

```
D2[ n_, k_ ] := Sum[ D2[ n/j, k-1 ], { j, 2, Floor[ n ] } ];D2[ n_, 0 ] := 1
Dd[x_, k_, y_] := Sum[ Dd[x/(j+y),k-1,y],{j,0,Floor[x-y]}];Dd[x_, 0, y_] := 1
Cc[x_, k_, y_] := y^-k Dd[ x y^k, k, y+1]
Grid[Table[ {D2[ n, k ],Cc[n,k,1]}, { n, 1, 50 }, { k, 1, 7 } ]]
```

On the other hand, if we take the limit as *y* approaches infinity, for the first few values of *k*, we have

$$\lim_{y\to\infty} C_{2,y}(x)=\int_1^x \int_1^{\frac{x}{z}} dw\, dz=x\log x-x+1=(1-\frac{\Gamma(2,-\log x)}{\Gamma(2)})$$

(7.26)

```
Dd[x_,0,y_]:=1; Dd[x_,1,y_]:=Floor[x]-y+1
Dd[x_,k_,y_]:=Sum[Binomial[k,j] Dd[x/(m^(k-j)),j,m+1],{m,y,x^(1/k)},{j,0,k-1}]
Cc[x_,k_,y_]:=y^-k Dd[x y^k,k,y+1]
Table[{Cc[x,2,3000.],N[x Log[x]-x+1],1-Gamma[2.,-Log[x]]/Gamma[2]},{x,2,40}]//TableForm
```

$$\lim_{y\to\infty} C_{3,y}(x)=\int_1^x \int_1^{\frac{x}{u}} \int_1^{\frac{x}{u\cdot z}} dw\, dz\, du=\frac{x}{2}(\log x)^2-x\log x+x-1=-(1-\frac{\Gamma(3,-\log x)}{\Gamma(3)})$$

(7.27)

```
Dd[x_,0,y_]:=1; Dd[x_,1,y_]:=Floor[x]-y+1
Dd[x_,k_,y_]:=Sum[Binomial[k,j] Dd[x/(m^(k-j)),j,m+1],{m,y,x^(1/k)},{j,0,k-1}]
Cc[x_,k_,y_]:=y^-k Dd[x y^k,k,y+1]
Table[{Cc[x,3,600.],N[x/2 Log[x]^2-x Log[x]+x-1],-(1-Gamma[3.,-Log[x]]/Gamma[3])},{x,2,10}]//TableForm
```

and, more generally, mirroring (7.9),

$$\boxed{\lim_{y\to\infty} C_{k,y}(x)=(-1)^k(1-\frac{\Gamma(k,-\log x)}{\Gamma(k)})}$$

(7.28)

```
Dd[x_,0,y_]:=1; Dd[x_,1,y_]:=Floor[x]-y+1
Dd[x_,k_,y_]:=Sum[Binomial[k,j] Dd[x/(m^(k-j)),j,m+1],{m,y,x^(1/k)},{j,0,k-1}]
Cc[x_,k_,y_]:=y^-k Dd[x y^k,k,y+1]
Table[{Cc[x,k,200.],N[(-1)^k(1-Gamma[k,-Log[x]]/Gamma[k])]},{x,2,7},{k,1,4}]//TableForm
```

where $\Gamma(k,-\log x)$ is the upper incomplete gamma function.

Following (7.7), we can express $D_{k,2}(x)$ as $C_{k,1}(x)=(\lim_{y\to\infty} C_{k,y}(x))-\int_1^{\infty} \frac{\partial}{\partial y} C_k(x,y)\,dy$, which is to say,

$$D_{k,2}(x)=(-1)^k(1-\frac{\Gamma(k,-\log x)}{\Gamma(k)})-\int_1^\infty \frac{\partial}{\partial y}(y^{-k}D_{k,y+1}(x y^k))dy$$

(7.29)

Now we can use (7.29) for identities for $D_z(n)$ and $\Pi(n)$.

If we apply (7.29) to (D1), $D_z(n)=\sum_{k=0}^{\lfloor\log_2 n\rfloor}\binom{z}{k}D_{k,2}(n)$, we have the following identity for $D_z(n)$

$$D_z(n)=\sum_{k=0}^{\infty}\binom{z}{k}((-1)^k(1-\frac{\Gamma(k,-\log n)}{\Gamma(k)})-\int_1^\infty \frac{\partial}{\partial y}(y^{-k}D_{k,y+1}(n y^k))dy)$$

(D12)

And, in like fashion, if we apply (7.29) to (P3), $\Pi(n)=\sum_{k=1}^{\lfloor\log_2 n\rfloor}\frac{(-1)^{k+1}}{k}D_{k,2}(n)$, we have this identity for the Riemann Prime counting function

$$\Pi(n)=\sum_{k=0}^{\infty}\frac{(-1)^{k-1}}{k}((-1)^k(1-\frac{\Gamma(k,-\log n)}{\Gamma(k)})-\int_1^\infty \frac{\partial}{\partial y}(y^{-k}D_{k,y+1}(n y^k))dy)$$

(7.30)

Finally, if we take advantage of the following identity for the logarithmic integral,

$$\sum_{k=1}^{\infty}\frac{(-1)^{k+1}}{k}\cdot((-1)^k(1-\frac{\Gamma(k,-\log n)}{\Gamma(k)}))=li(n)-\log\log n-\gamma$$

(7.31)

```
Table[ {N[Sum[ (-1)^(k+1)/k ( (-1)^k( 1- Gamma[ k,-Log[n]]/Gamma[k])),{k,1,Infinity}]], N[LogIntegral[n]-Log[Log[n]]-EulerGamma]}, {n,
100, 600, 100}]
```

then we are at last left with an expression connecting Riemann's Prime Counting function, $\Pi(n)$, with the logarithmic integral, $li(n)$.

$$\Pi(n)=li(n)-\log\log n-\gamma-\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{k}\int_1^\infty \frac{\partial}{\partial y}(y^{-k}D_{k,y+1}(n y^k))dy$$

(P12)

As another way of stating the general idea here, using $C_{k,y}(x)$ from (7.24), looking at the difference between $\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{k}C_{k,1}(x)$ and $\lim_{y\to\infty}\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{k}C_{k,y}(x)$ amounts to comparing

$$\Pi(n) = \sum_{j=2}^{\lfloor n\rfloor}1 - \frac{1}{2}\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor \frac{n}{j}\rfloor}1 + \frac{1}{3}\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor \frac{n}{j}\rfloor}\sum_{l=2}^{\lfloor \frac{n}{j\cdot k}\rfloor}1 - \frac{1}{4}\sum_{j=2}^{\lfloor n\rfloor}\sum_{k=2}^{\lfloor \frac{n}{j}\rfloor}\sum_{l=2}^{\lfloor \frac{n}{j\cdot k}\rfloor}\sum_{m=2}^{\lfloor \frac{n}{j\cdot k\cdot l}\rfloor}1 + \ldots$$

$$li(n) - \log\log n - \gamma = \int_1^n dx - \frac{1}{2}\int_1^n\int_1^{\frac{n}{x}} dy\,dx + \frac{1}{3}\int_1^n\int_1^{\frac{n}{x}}\int_1^{\frac{n}{xy}} dz\,dy\,dx - \frac{1}{4}\int_1^n\int_1^{\frac{n}{x}}\int_1^{\frac{n}{xy}}\int_1^{\frac{n}{xyz}} dw\,dz\,dy\,dx + \ldots$$

(7.32)

Some of the identities glossed over in this section are covered in more detail in http://www.icecreambreakfast.com/primecount/ApproximingThePrimeCountingFunctionWithLinniksIdentity_NathanMcKenzie.pdf

---

Applying a similar approach to the Chebyshev function, $\psi(n)=\sum_{j=2}^{\lfloor n\rfloor}\Lambda(j)$, if we can start with the function

$$L_{1,y}(x)=\sum_{j=0}^{\lfloor x-y\rfloor}\log\left(\frac{j+y}{y-1}\right);\; L_{k,y}(x)=\sum_{j=0}^{\lfloor x-y\rfloor}L_{k-1,y}\left(\frac{x}{j+y}\right)$$

(7.33)

```
l[x_, k_, y_] := Sum[ l[x/(j+y),k-1,y],{j,0,Floor[x-y]}]; l[x_, 1, y_] := Sum[ Log[(j+y)/(y-1)],{j,0,x-y}]; l[x_, 0, y_] := 1
```

we can then define the identity

$$C_{k,y}(n)=\sum_{k=1} y^{-1}(-1)^{k+1}L_{k,y+1}(n\,y^k)$$

which, for us, has two useful values. The first is

$$C_{k,1}(n)=\psi(n)$$

(7.34)

```
referenceChebyshev[n_]:=Sum[MangoldtLambda[j],{j,2,n}]
l[x_,k_,y_]:=Sum[l[x/(j+y),k-1,y],{j,0,Floor[x-y]}];l[x_,1,y_]:=Sum[Log[(j+y)/(y-1)],{j,0,x-y}];l[x_,0,y_]:=1
Cc[n_, y_]:=Sum[y^-1 (-1)^(k+1) l[n ,k,y+1],{k,1,Log[2,n]}]
Table[{n,N[referenceChebyshev[n]],N[Cc[n,1]]},{n,2,100}]//TableForm
```

the Chebyshev function. The second is

$$\lim_{y \to \infty} C_{k,y}(n) = n - \log n - 1$$

Consequently, we can express the difference between them as

$$\psi(n) = n - \log n - 1 - \sum_{k=1}^{\infty} (-1)^{k+1} \int_1^{\infty} \frac{\partial}{\partial y} \left(y^{-k} L_{k,y+1}(n y^k) \right) dy \tag{7.35}$$

This same general idea can also be expressed as

$$\begin{array}{ll} \psi(n) & = \sum_{j=2}^{n} \log j - \sum_{j=2}^{n} \sum_{k=2}^{\lfloor \frac{n}{j} \rfloor} \log j + \sum_{j=2}^{n} \sum_{k=2}^{\lfloor \frac{n}{j} \rfloor} \sum_{l=2}^{\lfloor \frac{n}{j \cdot k} \rfloor} \log j - \sum_{j=2}^{n} \sum_{k=2}^{\lfloor \frac{n}{j} \rfloor} \sum_{l=2}^{\lfloor \frac{n}{j \cdot k} \rfloor} \sum_{m=2}^{\lfloor \frac{n}{j \cdot k \cdot l} \rfloor} \log j + \ldots \\ n - \log n - 1 & = \int_1^{n} \log x \, dx - \int_1^{n} \int_1^{\frac{n}{x}} \log x \, dy \, dx + \int_1^{n} \int_1^{\frac{n}{x}} \int_1^{\frac{n}{x y}} \log x \, dz \, dy \, dx - \int_1^{n} \int_1^{\frac{n}{x}} \int_1^{\frac{n}{x y}} \int_1^{\frac{n}{x y z}} \log x \, dw \, dz \, dy \, dx + \ldots \end{array} \tag{7.36}$$

Some identities glossed over here are covered in more detail in http://www.icecreambreakfast.com/primecount/ApproximingThePrimeCountingFunctionWithLinniksIdentity_NathanMcKenzie.pdf

8. The Difference between $li(n)$ and $\Pi(n)$: A Partial Sum Equivalence to the Dirichlet Eta Function

This section shows another way to express the exact difference between $\Pi(n)$ and the logarithmic integral $li(n)$, here mirroring certain relationships between the Riemann Zeta function $\zeta(s)$ and the Dirichlet eta function $\eta(s)$, but in a partial sum, Dirichlet convolution context. As operations on zeta functions are generally more familiar, we'll work through the ideas in the section using $\zeta(s)$ to show the relationships, then use those same techniques on $D_z(n)$. The section will then end with a similar approach shown for connecting the Chebyshev function $\psi(n)$ and n.

Here's the approach we're going to take. Because the Zeta function and Dirichlet's eta function are more familiar than their Dirichlet convolution equivalents, and because they support normal mathematical operations (specifically being raised to complex powers), unlike their convolution equivalents, we'll work through the ideas here on them first. Then, we'll repeat those same identities and transformations in a convolution context (which is our actual goal). In the first part, we'll largely put aside concerns about questions of convergence.

So let's begin with the Dirichlet eta function.
The Dirichlet eta function $\eta(s)$ is defined as

$$\eta(s)=\sum_{n=1}^{\infty}\frac{(-1)^{n-1}}{n^s} \tag{8.1}$$

```
eta[s_] := Sum[ (-1)^(n-1)/n^s,{n,1,Infinity}]
Table[ eta[s],{s,1,6}]
```

It is well known that it can be expressed in terms of the Riemann Zeta function $\zeta(s)$ as

$$\eta(s)=(1-2^{1-s})\zeta(s) \tag{8.2}$$

```
eta[s_] := Sum[ (-1)^(n-1)/n^s,{n,1,Infinity}]
{Expand[eta[s]],Expand[(1-2^(1-s))Zeta[s]]}
```

which, inverted, gives us

$$\zeta(s)=(1-2^{1-s})^{-1}\eta(s) \tag{8.3}$$

```
eta[s_]:=Sum[(-1)^(n-1)/n^s,{n,1,Infinity}]
```

```
{FullSimplify[(1-2^(1-s))^(-1) eta[s]],Zeta[s]}
```

If $\Re(s)>1$, we can use a variant of the binomial theorem, $(1-a)^{-1}=\sum_{j=0}^{\infty}(-1)^j\binom{-1}{j}a^j$,

```
{Sum[(-1)^j Binomial[-1,j]a^j,{j,0,Infinity}],(1-a)^-1}
```

to rewrite (8.3) as

$$\zeta(s)=\sum_{j=0}^{\infty}(-1)^j\binom{-1}{j}(2^{1-s})^j\eta(s) \tag{8.4}$$

```
eta[s_] :=(1-2^(1-s))Zeta[s]
{Zeta[s],Sum[ (-1)^j Binomial[-1,j] (2^(1-s))^j eta[s],{j,0,Infinity}]}
```

which generalizes, for $\zeta(s)^z$, to

$$\zeta(s)^z=\sum_{j=0}^{\infty}(-1)^j\binom{-z}{j}(2^{1-s})^j\eta(s)^z \tag{8.5}$$

```
eta[s_] :=(1-2^(1-s))Zeta[s]
Table[{Zeta[s]^z,FullSimplify[Sum[ (-1)^j Binomial[-z,j] (2^(1-s))^j eta[s]^z,{j,0,Infinity}]]},{z,-4,4}]//TableForm
```

Now, anywhere $0<\eta(s)<2$, we can also express $\eta(s)^z$ using the generalized binomial theorem,

$$\eta(s)^z=\sum_{k=0}^{\infty}\binom{z}{k}(\eta(s)-1)^k \tag{8.6}$$

```
eta[s_] :=(1-2^(1-s))Zeta[s]
FullSimplify[{eta[s]^z,Sum[ Binomial[z,k](eta[s]-1)^k,{k,0,Infinity}]}]
```

so we can express $\zeta(s)^z$ as

$$\zeta(s)^z=\sum_{j=0}^{\infty}\sum_{k=0}^{\infty}(-1)^j\binom{-z}{j}\binom{z}{k}(2^{1-s})^j(\eta(s)-1)^k \tag{8.7}$$

```
eta[s_] :=(1-2^(1-s))Zeta[s]
Table[ { Zeta[s]^z, FullSimplify[Sum[ (-1)^j Binomial[-z,j]Binomial[z,k](2^(1-s))^j (eta[s]-1)^k,{j,0,Infinity},{k,0,Infinity}]]}, {z,-3,3}]//TableForm
```

Let's generalize $\eta(s)$ to $\eta(s,c)$, where c is a rational number of the form $\frac{a}{b}$ that replaces 2 in our previous equations, giving us

$$\eta(s,c)=(1-c^{1-s})\zeta(s) \tag{8.8}$$

```
eta[s_, c_] :=(1-c^(1-s))Zeta[s]
```

and apply that to our expression for $\zeta(s)^z$

$$\zeta(s)^z = \sum_{j=0}^{\infty}\sum_{k=0}^{\infty}(-1)^j\binom{-z}{j}\binom{z}{k}(c^{1-s})^j(\eta(s,c)-1)^k \tag{8.9}$$

```
eta[s_, c_] :=(1-c^(1-s))Zeta[s]
Table[ { Zeta[s]^z, FullSimplify[Sum[ (-1)^j Binomial[-z,j]Binomial[z,k](c^(1-s))^j (eta[s,c]-1)^k,{j,0,Infinity},{k,0,Infinity}]]}, {z,-3,3}]//TableForm
```

Now apply the identity $\log x = \lim_{z\to 0}\frac{x^z-1}{z}$ to arrive at

$$\lim_{z\to 0}\frac{\zeta(s)^z-1}{z} = \lim_{z\to 0}\frac{1}{z}\left(-1+\sum_{j=0}^{\infty}\sum_{k=0}^{\infty}(-1)^j\binom{-z}{j}\binom{z}{k}(c^{1-s})^j(\eta(s,c)-1)^k\right) \tag{8.10}$$

```
eta[s_, c_] :=(1-c^(1-s))Zeta[s]
Table[{FullSimplify[Limit[(Zeta[s]^z-1)/z,z->0]],FullSimplify[Limit[ (Sum[ (-1)^j Binomial[-z,j]Binomial[z,k](c^(1-s))^j (eta[s,c]-1)^k,{j,0,Infinity},{k,0,Infinity}]-1)/z,z->0]]},{s,2,5}]
```

The coefficients produced by the product $\lim_{z\to 0}(-1)^j\frac{1}{z}\binom{-z}{j}\binom{z}{k}$ resolve into four kinds of terms. For both *k* and *j* non-zero, the product is 0. When both *j* and *k* are 0, the limit goes to infinity, but is canceled by the -1. For *k* non-zero and *j* 0, the coefficient is $\frac{(-1)^{k-1}}{k}$. For *j* non-zero and *k* 0, the coefficient is $\frac{1}{j}$.

```
Grid[Table[ Limit[ (-1)^j 1/z Binomial[-z,j] Binomial[z,k],z->0],{j,0,7},{k,0,7}]]
```

So, taking the log of our expression for $\zeta(s)^z$ in (8.9), we have

$$\log\zeta(s) = \sum_{j=1}^{\infty}\frac{(c^{1-s})^j}{j}+\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{k}(\eta(s,c)-1)^k \tag{8.11}$$

```
eta[s_, c_] :=(1-c^(1-s))Zeta[s]
Table[{Log[Zeta[s]],FullSimplify[Sum[ (c^(1-s))^j/j,{j,1,Infinity}]+Sum[ (-1)^(k-1)/k( eta[s,c]-1)^k, {k,1,Infinity}]]}, {s,2,5},{x,2,5}]
```

Now, let $c=\frac{b+1}{b}$ and take the limit as *b* approaches infinity:

$$\log\zeta(s)=\lim_{b\to\infty}\sum_{j=1}^{\infty}\frac{\left(\left(\frac{b+1}{b}\right)^{1-s}\right)^{j}}{j}+\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{k}\left(\eta\left(s,\frac{b+1}{b}\right)-1\right)^{k}$$

(8.12)

```
eta[s_, c_] :=(1-c^(1-s))Zeta[s]
Table[{Log[Zeta[s]],Limit[ Sum[ (((b+1)/b)^(1-s))^j/j,{j,1,Infinity}]+Sum[ (-1)^(k-1)/k( eta[s,(b+1)/b]-1)^k, {k,1,Infinity}],b->Infinity]},
{s,2,5}]
```

and then hope there's some interesting manipulation we can do to these last sums to provide insight. That's the basic mechanics of what we're about to do in a partial sum context.

These same relationships largely hold in a partial sum context.

$D_z(n)$ will map to $\zeta(s)^z$, from (8.5).

$\Pi(n)$ will map to $\log\zeta(s)$, from (8.11).

Mapping to $\eta(s)^z$, from (8.5), is

$$E_0(n)=1;\ E_k(n)=\sum_{j=1}^{\lfloor n\rfloor}(-1)^{j+1}E_{k-1}\left(\frac{n}{j}\right)$$

(8.13)

```
E1[n_,k_]:= Sum[ (-1)^(j+1)E1[n/j,k-1],{j,1,n}];E1[n_,0]:=1
Table[ E1[ n, k ], { n, 1, 50 }, { k, 1, 7 } ] // TableForm
```

and mapping to $(\eta(s)-1)^k$, from (8.7), is

$$E_{0,2}(n)=1;\ E_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor}(-1)^{j+1}E_{k-1,2}\left(\frac{n}{j}\right)$$

(8.14)

```
E2[n_,k_]:= Sum[ (-1)^(j+1)E2[n/j,k-1],{j,2,n}];E2[n_,0]:=1
Table[ E2[ n, k ], { n, 1, 50 }, { k, 1, 7 } ] // TableForm
```

Already we can use this to see that

$$\Pi(n)=\sum_{j=1}^{\lfloor\log_2 n\rfloor}\frac{2^j}{j}+\sum_{k=1}^{\lfloor\log_2 n\rfloor}\frac{(-1)^k}{k}E_{k,2}(n)$$

```
referenceRiemanPrimeCount[n_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]],{j,2,n}]
E2[n_,k_]:= Sum[ (-1)^(j+1)E2[n/j,k-1],{j,2,n}];E2[n_,0]:=1
RiemanPrimeCountAlt[n_] := Sum[ 2^j/j,{j,1,Log[2,n]}]+Sum[ (-1)^(k+1)/k E2[n,k],{k,1,Log[2,n]}]
```

```
Table[ {referenceRiemanPrimeCount[ n ], RiemanPrimeCountAlt[n] }, { n, 1,100 } ] // TableForm
```

Expanded out, this identity can also be written as

$$\Pi(n)=\sum_{j=1}^{\lfloor \log_2 n \rfloor} \frac{2^j}{j}+\sum_{j=2}^{n}(-1)^{j+1}-\frac{1}{2}\sum_{j=2}^{n}\sum_{k=2}^{\lfloor \frac{n}{j} \rfloor}(-1)^{j+1}\cdot(-1)^{k+1}+\frac{1}{3}\sum_{j=2}^{n}\sum_{k=2}^{\lfloor \frac{n}{j} \rfloor}\sum_{l=2}^{\lfloor \frac{n}{j\cdot k} \rfloor}(-1)^{j+1}\cdot(-1)^{k+1}\cdot(-1)^{l+1}-\frac{1}{4}...$$

Now we'll generalize this notion of alternating series to the rationals. Essentially, we replace the function $(-1)^{n+1}$ with the function

$$\alpha(n,\frac{a}{b})=b\cdot(\lfloor \frac{n}{b} \rfloor-\lfloor \frac{n-1}{b} \rfloor)-a\cdot(\lfloor \frac{n}{a} \rfloor-\lfloor \frac{n-1}{a} \rfloor)$$

where a and b are the numerator and denominator of some rational number c. You can verify that $\alpha(n,\frac{2}{1})=(-1)^{n+1}$.

```
num[c_]:=Numerator[c];den[c_]:=Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
Table[{n,alpha[n,2],(-1)^(n+1)},{n,1,50}]//TableForm
```

(As an aside, $\alpha(n,\frac{a}{b})$ can also be used to generalize the very well known $\sum_{n=1}^{\infty}\frac{(-1)^{j+1}}{n}=\log 2$ to

$$\sum_{n=1}^{\infty}\frac{\alpha(n,\frac{a}{b})}{n}=\log\frac{a}{b}$$

```
num[c_]:=Numerator[c];den[c_]:=Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
Grid[Table[{Sum[N[alpha[n,a/b]/n],{n,1,100000}],N[Log[a/b]]},{a,1,10},{b,1,6}]]
```

)

At any rate, continuing the pattern laid out previously with the zeta function, we can mirror the generalization of $\eta(s,\frac{a}{b})^z$ from (8.8) with the following function, where c is some rational constant fraction of the form $c=\frac{a}{b}, a>b$. Then

$$E_0(n)=1;\ E_k(n)=\frac{1}{b}\sum_{j=1}^{\lfloor n\cdot b \rfloor}\alpha(j,c)E_{k-1}(\frac{n\cdot b}{j}) \tag{8.15}$$

```
num[c_] := Numerator[c]; den[c_] := Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
E1[n_,k_,c_]:=(1/den[c])Sum[If[alpha[j,c] == 0,0,alpha[j,c]E1[(den[c] n)/j,k-1,c]],{j,1,den[c] n}];E1[n_,0,c_]:=1
```

and corresponding to $(\eta(s,\frac{a}{b})-1)^k$, from (8.9), is

$$E_{0,2}(n)=1;\ E_{k,2}(n)=\frac{1}{b}\sum_{j=b+1}^{\lfloor n\cdot b\rfloor}\alpha(j,c)E_{k-1,2}\left(\frac{n\cdot b}{j}\right) \tag{8.16}$$

```
num[c_] := Numerator[c]; den[c_] := Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
E2[n_,k_,c_]:=(1/den[c])Sum[If[alpha[j,c] == 0,0,alpha[j,c]E2[(den[c] n)/j,k-1,c]],{j,den[c]+1,den[c] n}];E2[n_,0,c_]:=1
```

Using the general approach from section 5, both of these can be calculated more quickly with

$$F_{k,s}(n)=\sum_{j=1}^{k}\binom{k}{j}\sum_{m=s}^{\lfloor n^{\frac{1}{k}}\rfloor}\alpha\left(m,\frac{a}{b}\right)^{j}F_{k-j,m+1}\left(\frac{n}{m^{j}}\right)$$

$$F_{1,s}(n)=\left(b\left\lfloor\frac{n}{b}\right\rfloor-a\left\lfloor\frac{n}{a}\right\rfloor\right)-\left(b\left\lfloor\frac{(s-1)}{b}\right\rfloor-a\left\lfloor\frac{(s-1)}{a}\right\rfloor\right)$$

$$F_{0,s}(n)=1$$

$$E_k(n)=b^{-k}F_{k,1}(n b^{k})$$

$$E_{k,2}(n)=b^{-k}F_{k,b+1}(n b^{k})$$

```
num[c_]:=Numerator[c];den[c_]:=Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
F[n_,0,s_, c_]:=1
F[ n_, 1, s_, c_] := If[ n < s, 0, (den[c] Floor[n/den[c]]-num[c] Floor[n/num[c]])-(den[c] Floor[(s-1)/den[c]]-num[c] Floor[(s-1)/num[c]])]
F[n_,k_,s_, c_]:=F[n,k,s,c]=Sum[If[ alpha[m,c] ⊠0, 0, Binomial[k,j] alpha[m,c]^j F[Floor[n/(m^j)],k-j,m+1, c]],{j,1,k},{m,s,Floor[n^(1/k)]}]
E1Alt[n_, k_, c_] := den[c]^-k F[ n den[c]^k, k, 1, c]
E2Alt[n_, k_, c_] := den[c]^-k F[ n den[c]^k, k, den[c]+1, c]
E1[n_,k_,c_]:=(1/den[c])Sum[If[alpha[j,c]⊠0,0,alpha[j,c]E1[(den[c] n)/j,k-1,c]],{j,1,den[c] n}];E1[n_,0,c_]:=1
E2[n_,k_,c_]:=(1/den[c])Sum[If[alpha[j,c]⊠0,0,alpha[j,c]E2[(den[c] n)/j,k-1,c]],{j,den[c]+1,den[c] n}];E2[n_,0,c_]:=1
Grid[Table[{E1[n,3,(b+1)/b],E1Alt[n,3,(b+1)/b]},{n,10,80,10},{b,1,6}]]
Grid[Table[{E2[n,3,(b+1)/b],E2Alt[n,3,(b+1)/b]},{n,10,80,10},{b,1,6}]]
```

With these definitions in place, a similar approach, the details of which we'll skip over here, gives us the following.

As before, c is some rational constant fraction of the form $c=\frac{a}{b}, a>b$.

Filling the role, roughly, of (8.6), we can express (8.15) in terms of (8.16) as

$$\boxed{E_z(n)=\sum_{k=0}^{\lfloor\frac{\log n}{\log c}\rfloor}\binom{z}{k}E_{k,2}(n)} \tag{8.18}$$

```
num[c_] := Numerator[c]; den[c_] := Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
E1[n_,k_,c_]:=E1[n,k,c]=(1/den[c])Sum[If[alpha[j,c] == 0,0,alpha[j,c]E1[(den[c] n)/j,k-1,c]],{j,1,den[c] n}];E1[n_,0,c_]:=1
E2[n_,k_,c_]:=E2[n,k,c]=(1/den[c])Sum[If[alpha[j,c] == 0,0,alpha[j,c]E2[(den[c] n)/j,k-1,c]],{j,den[c]+1,den[c] n}];E2[n_,0,c_]:=1
```

```
E1Alt[n_,z_,c_]:=Sum[Binomial[z,k] E2[n,k,c],{k,0,Floor[Log[n]/Log[c]]}]
Grid[Table[{E1[n,3,(b+1)/b],E1Alt[n,3,(b+1)/b]},{n,10,80,10},{b,1,6}]]
```

Our function from (8.15) can be expressed in terms of the generalized divisor summatory function as

$$E_z(n)=\sum_{j=0}^{\lfloor\frac{\log n}{\log c}\rfloor}(-1)^j\binom{z}{j}c^j D_z\left(\frac{n}{c^j}\right)$$

(8.17)

```
d1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
ReferenceD1[n_,z_]:=Sum[d1[j,z],{j,1,n}]
num[c_]:=Numerator[c];den[c_]:=Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
E2[n_,k_,c_]:=E2[n,k,c]=(1/den[c])Sum[If[alpha[j,c]==0,0,alpha[j,c]E2[(den[c] n)/j,k-1,c]],{j,den[c]+1,den[c] n}];E2[n_,0,c_]:=1
E1[n_,z_,c_]:=Sum[Binomial[z,k] E2[n,k,c],{k,0,Floor[Log[n]/Log[c]]}]
E1Alt[n_,z_,c_]:=Sum[(-1)^j Binomial[z,j] c^j ReferenceD1[n/c^j,z],{j,0,Floor[Log[n]/Log[c]]}]
Grid[Table[{E1[a=111,s+t I, 4/3],E1Alt[a,s+t I, 4/3]},{s,-1.3,4,.7},{t,-1.3,4,.7}]]
```

and its inverse, a relationship similar to that of (8.5), expresses $D_z(n)$ as

$$D_z(n)=\sum_{j=0}^{\lfloor\frac{\log n}{\log c}\rfloor}(-1)^j\binom{-z}{j}c^j E_z\left(\frac{n}{c^j}\right)$$

(D13)

```
d1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
ReferenceD1[n_,z_]:=Sum[d1[j,z],{j,1,n}]
num[c_] := Numerator[c]; den[c_] := Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
E2[n_,k_,c_]:=E2[n,k,c]=(1/den[c])Sum[If[alpha[j,c] == 0,0,alpha[j,c]E2[(den[c] n)/j,k-1,c]],{j,den[c]+1,den[c] n}];E2[n_,0,c_]:=1
E1[n_,z_,c_]:=Sum[Binomial[z,k] E2[n,k,c],{k,0,Floor[Log[n]/Log[c]]}]
D1Alt[n_,z_,c_]:=Sum[(-1)^j Binomial[-z,j] c^j E1[n/c^j,z,c],{j,0,Floor[Log[n]/Log[c]]}]
Grid[Table[{ReferenceD1[a=111,s+t I],D1Alt[a,s+t I,5/4]},{s,-1.3,4,.7},{t,-1.3,4,.7}]]
```

And finally, corresponding to (8.9) is this second identity for the generalized divisor summatory function $D_z(n)$ in terms of $E_{k,2}(n)$ from (8.16)

$$D_z(n)=\sum_{j=0}^{\lfloor\frac{\log n}{\log c}\rfloor}\sum_{k=0}^{\lfloor\frac{\log n+j\log c}{\log c}\rfloor}(-1)^j\binom{-z}{j}\binom{z}{k}c^j E_{k,2}\left(\frac{n}{c^j}\right)$$

(D14)

```
d1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
ReferenceD1[n_,z_]:=Sum[d1[j,z],{j,1,n}]
num[c_] := Numerator[c]; den[c_] := Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
E2[n_,k_,c_]:=E2[n,k,c]=(1/den[c])Sum[If[alpha[j,c] == 0,0,alpha[j,c]E2[(den[c] n)/j,k-1,c]],{j,den[c]+1,den[c] n}];E2[n_,0,c_]:=1
D1Alt[n_,z_,c_]:=Sum[(-1)^j Binomial[-z,j] Binomial[z,k] c^j  E2[n /c^j,k,c],{j,0,Floor[Log[n]/Log[c]]},{k,0,Floor[(Log[n]-j
Log[c])/Log[c]]}]
Grid[Table[{ReferenceD1[123,j+1/3],D1Alt[123,j+1/3,(b+1)/b]},{j,1,5},{b,1,5}]]
```

The pair of binomials in (D14) simplify when z is an integer. In particular, if z is 2, we have the Dirichlet Divisor function $D(n)$, and if z is -1, we have the Mertens function $M(n)$.

$$D(n)=\sum_{j=0}^{\lfloor \frac{\log n}{\log c} \rfloor} (j+1)c^j \left(E_{0,2}\left(\frac{n}{c^j}\right)+2E_{1,2}\left(\frac{n}{c^j}\right)+E_{2,2}\left(\frac{n}{c^j}\right)\right)$$

$$M(n)=\sum_{k=0}^{\lfloor \frac{\log n}{\log c} \rfloor} (-1)^k \left(E_{k,2}(n)-c\,E_{k,2}\left(\frac{n}{c}\right)\right)$$

```
d1[n_,z_]:=Product[(-1)^p[[2]] Binomial[-z,p[[2]]],{p,FI[n]}];FI[n_]:=FactorInteger[n];FI[1]:={}
ReferenceD1[n_,z_]:=Sum[d1[j,z],{j,1,n}]
MertensReference[n_]:=Sum[MoebiusMu[j],{j,1,n}]
num[c_] := Numerator[c]; den[c_] := Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
E2[n_,k_,c_]:=E2[n,k,c]=(1/den[c])Sum[If[alpha[j,c] == 0,0,alpha[j,c]E2[(den[c] n)/j,k-1,c]],{j,den[c]+1,den[c] n}];E2[n_,0,c_]:=1
DAlt[n_,c_]:=Sum[(j+1)c^j (E2[n /c^j,0,c]+2 E2[n /c^j,1,c]+E2[n /c^j,2,c]),{j,0,Log[n]/Log[c]}]
MertensAlt[n_,c_]:=Sum[(-1)^k (E2[n,k,c]-c E2[n /c,k,c]),{k,0,Floor[Log[n]/Log[c]]}]
Grid[Table[{ReferenceD1[n,2],DAlt[n,(b+1)/b]},{n,10,100,10},{b,1,7}]]
Grid[Table[{MertensReference[n],MertensAlt[n,(b+1)/b]},{n,10,100,10},{b,1,5}]]
```

The general hope for these last 3 equations would be to take the limit as $c \to 1^+$ and then transform the remaining sums in such a way that something interesting can be said about them. Although no such approach is illustrated here for $D(n)$ or $M(n)$, we can take exactly that approach for $\Pi(n)$, which follows below.

Continuing with these parallel identities from the beginning of this section, mapping to (8.11), if we start with (D14) and then use the identity $\Pi(n)=\lim_{z\to 0}\frac{D_z(n)-1}{z}$, then the Riemann Prime counting function $\Pi(n)$ can be expressed in terms of $E_{k,2}(n)$, from (8.16), as

$$\Pi(n)=\sum_{j=1}^{\lfloor \frac{\log n}{\log c} \rfloor} \frac{c^j}{j}+\sum_{k=1}^{\lfloor \frac{\log n}{\log c} \rfloor} \frac{(-1)^k}{k} E_{k,2}(n) \tag{P13}$$

```
referenceRiemanPrimeCount[n_]:=Sum[PrimePi[n^(1/k)]/k,{k,1,Floor[Log[2,n]]}]
num[c_] := Numerator[c]; den[c_] := Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
E2[n_,k_,c_]:=E2[n,k,c]=(1/den[c])Sum[If[alpha[j,c] == 0,0,alpha[j,c]E2[(den[c] n)/j,k-1,c]],{j,den[c]+1,den[c] n}];E2[n_,0,c_]:=1
P[n_,c_]:=Sum[c^j/j,{j,1,Floor[Log[n]/Log[c]]}]+Sum[(-1)^(k+1)/k E2[n,k,c],{k,1,If[c<2,Floor[Log[n]/Log[c]],Log[2,n]]}]
Table[{n,referenceRiemanPrimeCount[n],P[n,5/2],P[n,3/2],P[n,4/3]},{n,1,100}]//TableForm
```

Now, it can be shown that

$$\lim_{c\to 1^+}\sum_{j=1}^{\lfloor \frac{\log n}{\log c} \rfloor} \frac{c^j-1}{j}=li(n)-\log\log n-\gamma$$

(8.19)

```
Table[{n,Sum[N[(c^j-1)/j],{j,1,Floor[Log[n]/Log[c]]}]/.c->1.00001,N[LogIntegral[n]-Log[Log[n]]-EulerGamma]},{n,5,100,5}]//TableForm
```

(which has a nice visual resemblance to the very well known $\lim_{j\to 0}\frac{c^j-1}{j}=\log c$), meaning that if we take the limit in (P13) as c approaches 1 from above, similar to what we did in (8.12), then we finally have an equation expressing the relationship between $\Pi(n)$ and the logarithmic integral $li(n)$. With $c=\frac{b+1}{b}$, we have

$$\Pi(n)=li(n)-\log\log n-\gamma+\lim_{b\to\infty}\sum_{k=1}^{\lfloor\frac{\log n}{\log(b+1)-\log b}\rfloor}\frac{(-1)^{k-1}}{k}E_{k,2}(n)+\frac{1}{k}$$

(P14)

A similar technique can be applied to the Chebyshev function, $\psi(n)=\sum_{j=2}^{n}\Lambda(j)$. If we define the following function, analogous to $E_{k,2}(n)$ from (8.16), again with c some rational constant fraction of the form $c=\frac{a}{b}, a>b$, then

$$L_{1,2}(n)=\sum_{j=b+1}^{\lfloor n\cdot b\rfloor}\alpha(j,c)\log\frac{j}{b};\ L_{k,2}(n)=\frac{1}{b}\sum_{j=b+1}^{\lfloor n\cdot b\rfloor}\alpha(j,c)L_{k-1,2}(\frac{n\cdot b}{j})$$

(8.20)

```
num[c_]:=Numerator[c];den[c_]:=Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
L2[n_,1,c_]:=L2[n,1,c]=(1/den[c])Sum[alpha[ j, c]Log[j/den[c]],{j,den[c]+1,den[c] n}]
L2[n_,k_,c_]:=L2[n,k,c]=(1/den[c])Sum[If[ alpha[j,c] == 0, 0, alpha[j,c]L2[den[c] n/j,k-1,c]],{j,den[c]+1,den[c] n}]
```

and we will find that $\psi(n)$ can be expressed as

$$\psi(n)=\sum_{k=1}^{\lfloor\frac{\log n}{\log c}\rfloor}(-1)^{k-1}L_{k,2}(n)+c^k\cdot\log c$$

(8.21)

```
referenceChebyshev[n_]:=Sum[MangoldtLambda[j],{j,2,n}]
num[c_]:=Numerator[c];den[c_]:=Denominator[c]
alpha[n_,c_]:=den[c] (Floor[n/den[c]]-Floor[(n-1)/den[c]])-num[c] (Floor[n/num[c]]-Floor[(n-1)/num[c]])
L2[n_,1,c_]:=L2[n,1,c]=(1/den[c])Sum[alpha[ j, c]Log[j/den[c]],{j,den[c]+1,den[c] n}]
L2[n_,k_,c_]:=L2[n,k,c]=(1/den[c])Sum[If[ alpha[j,c] ⌧0, 0, alpha[j,c]L2[den[c] n/j,k-1,c]],{j,den[c]+1,den[c] n}]
ChebAlt[n_,c_]:=Sum[(-1)^(k-1) L2[n,k,c],{k,1,Floor[ Log[n]/Log[If[c<2,c,2]]]}]+Sum[c^k Log[c],{k,1,Floor[Log[n]/Log[c]]}]
Grid[Table[ {N[ referenceChebyshev[n]], N[ChebAlt[n,(b+1)/b]] },{n,5,100,5},{b,1,4}]]
```

Now, given the following limit,

$$\lim_{c\to 1^+} \sum_{k=1}^{\lfloor \frac{\log n}{\log c} \rfloor} c^k \cdot \log c = n-1$$

(8.22)

```
{Limit[Sum[c^k Log[c],{k,1,Log[n]/Log[c]}],c->1],n-1}
```

this means that the relationship between $\psi(n)$ and n can be expressed, with $c=\frac{b+1}{b}$, as

$$\boxed{\psi(n)=n-1+\lim_{b\to\infty} \sum_{k=1}^{\lfloor \frac{\log n}{\log(b+1)-\log b} \rfloor} (-1)^{k-1} L_{k,2}(n)}$$

(8.23)

More Notes and Further Questions

This paper has been particularly concerned with Riemann's Prime counting function, $\Pi(n)$. However, many of the identities here for the generalized divisor summatory function $D_z(n)$ might also be relevant for the Mertens function, which is $D_{-1}(n)$, and the original Dirichlet Divisor Problem, $D_2(n)$. See all of the identities for $D_z(n)$ in Section 2, (D8), (D9), (4.1), (4.2), (D10), (D11), (4.4), (4.5), (5.6), (D12), (D13), and (D14).

For the sake of simplicity, this paper has only covered $D_{0,2}(n)=1;\ D_{k,2}(n)=\sum_{j=2}^{\lfloor n\rfloor} D_{k-1,2}(\frac{n}{j})$. However, just as the Riemann Zeta function has its parameter *s,* most of of the identities in this paper can be generalized to a more complicated function, $D_{0,2}(s,n)=1;\ D_{k,2}(s,n)=\sum_{j=2}^{\lfloor n\rfloor} j^{-s}\cdot D_{k-1,2}(s,\frac{n}{j})$. In particular, the approximation techniques in sections 7 and 8 accommodate this extra variable without great difficulty, all the identities in sections 2 and 3 have corresponding identities with this extra parameter, the role of roots in section 4 continue to work, and the techniques for computation from sections 5 and 6 function and, if *s* is a negative integer, also preserve their time and space performance characteristics. In particular, if *s* is -1, these techniques can be used to quickly sum primes – see http://www.icecreambreakfast.com/primecount/PrimeSumming_NathanMcKenzie.pdf for a description of how this generalization works, and `http://www.icecreambreakfast.com/primecount/primesumcount.cpp` for an implementation of Section 6's algorithm, adjusted to sum primes.

Although sections 7 and 8 successfully show connections between $\Pi(n)$ and $li(n)$, the logarithmic integral, they're often left in forms that are quite hard to reason about or work with, as are some other interesting identities for $D_z(n)$, $D_{k,2}(n)$, and $\psi(n)$. In particular, it would be ideal to have further development / simplification of (7.29), (D12), (P12), (7.34),(D14), (P14), and (8.23).

The method of using roots in section 4 seems interesting, but in its present form is little more than a gesture. A richer study of this approach could be useful.